

Κεφάλαια 2, 7, 8

Βασικές έννοιες αλγορίθμων - προγραμματισμού

(§2.1) Η έννοια Αλγόριθμος.

Κριτήρια αλγορίθμου:

1. **Είσοδος** (προαιρετική, π.χ. γεννήτρια τυχαίων αριθμών)
2. **Έξοδος**
3. **Καθοριστικότητα** (καμία αμφιβολία για τον τρόπο εκτέλεσης της κάθε εντολής π.χ. διαίρεση με το μηδέν)
4. **Περατότητα** (όταν παραβιάζεται, ο μηχανισμός ονομάζεται “υπολογιστική διαδικασία”)
5. **Αποτελεσματικότητα** (κάθε εντολή είναι απλή και εκτελέσιμη)

(§2.2) Σκοπιές μελέτης των αλγορίθμων από την Πληροφορική

- Υλικού ~ αρχιτεκτονική ΗΥ και ταχύτητα του αλγορίθμου (π.χ. RAM)
- Γλωσσών προγραμματισμού ~ ταχύτητα
- Θεωρητική (ποιος ο βέλτιστος αλγόριθμος;)
- Αναλυτική (απαιτήσεις του αλγορίθμου σε υπολογιστικούς πόρους: RAM, CPU, IO)

(§2.3) Τρόποι αναπαράστασης αλγορίθμων:

1. ελεύθερο κείμενο (π.χ. μαγειρική συνταγή) – ο λιγότερο δομημένος τρόπος ☒ αποτελεσματικότητα
2. διαγραμματικές τεχνικές
3. φυσική γλώσσα κατά βήματα (π.χ. manual) ☒ καθοριστικότητα
4. κωδικοποίηση

(τα 1, 2 και 3 απευθύνονται στον άνθρωπο, ενώ το 4 στον ΗΥ)

(§7.1) Στοιχεία της ΓΛΩΣΣΑΣ (αλφάβητο: γράμματα ελληνικά-λατινικά, ψηφία, ειδικά σύμβολα)

(§7.2) Τύποι δεδομένων:

1. Ακέραιος π.χ. 3, -9, 2004
2. Πραγματικός π.χ. 3.14
3. Χαρακτήρας π.χ. '3ο Γενικό Λύκειο Ηρακλείου'
4. Λογικός π.χ. έγγαμος

(§7.3) Σταθερές:

- α) σταθερές τιμές: τιμές οποιουδήποτε τύπου, που δεν αλλάζουν κατά την εκτέλεση του προγράμματος.
- β) συμβολικές σταθερές: ορισμός, τύποι, δήλωση, παραδείγματα, πλεονεκτήματα (πιο ευανάγνωστο και πιο ευέλικτο σε αλλαγές πρόγραμμα)

Σύνταξη

ΣΤΑΘΕΡΕΣ

Όνομα-1 = σταθερή-τιμή-1

Όνομα-2 = σταθερά-τιμή-2

.

.

.

Όνομα-ν = σταθερά-τιμή-ν

Παραδείγματα


ΣΤΑΘΕΡΕΣ

ΠΙ=3.14159

ΦΠΑ=0.18

ΟΝΟΜΑ=' Κώστας'

Διαφοροποίηση προγραμματιστή και χρήστη



Κόσμος προγραμματιστή

1. Απαιτείται να γνωρίζει προγραμματισμό (κατασκευή προγραμμάτων) – αποκτάται με σπουδές
2. Κατασκευάζει / συντηρεί / διορθώνει προγράμματα
3. Έχει πρόσβαση στον κώδικα και στη διεπαφή χρήσης του προγράμματος
4. Τον αφορούν οι έννοιες «ευανάγνωστο πρόγραμμα» και «φιλικότητα προς το χρήστη»


Κόσμος Χρήστη

1. Απαιτείται να γνωρίζει χειρισμό (χρήση προγραμμάτων) – αποκτάται με σεμινάρια χρήσης
2. Χειρίζεται προγράμματα
3. Έχει πρόσβαση ΜΟΝΟ στη διεπαφή χρήσης του προγράμματος (εκτελέσιμο)
4. Τον ΜΟΝΟ αφορά η έννοια «πρόγραμμα φιλικό προς το χρήστη»

(§7.4) Μεταβλητές: ορισμός, τύποι, δήλωση, παραδείγματα (λειτουργία του ΗΥ: δέσμευση χώρου στη μνήμη με **ΑΠΡΟΣΔΙΟΡΙΣΤΕΣ** τιμές),.

Αντιστοιχίζονται από το μεταγλωττιστή σε συγκεκριμένες θέσεις μνήμης επιτρεπτά ονόματα - κανόνες:

1. όλες οι μεταβλητές να δηλώνονται
2. όχι 2 μεταβλητές με το ίδιο όνομα
3. ονόματα σχετικά με τα δεδομένα → πιο ευανάγνωστο
4. όχι δεσμευμένες λέξεις
5. όχι κενά ενδιάμεσα
6. όχι αριθμοί στην αρχή
7. όχι ειδικά σύμβολα



Σύνταξη

ΜΕΤΑΒΛΗΤΕΣ

τύπος-1: Λίστα-μεταβλητών-1

τύπος-2: Λίστα-μεταβλητών-2

.

.

.

Τύπος-ν: Λίστα-μεταβλητών-ν

Παραδείγματα

ΜΕΤΑΒΛΗΤΕΣ

ΠΡΑΓΜΑΤΙΚΕΣ: Εμβαδόν, Α

ΑΚΕΡΑΙΕΣ: ΤΙΜΗ, Ν

ΧΑΡΑΚΤΗΡΕΣ: Όνομα

ΛΟΓΙΚΕΣ: Έλεγχος

(§7.5) Αριθμητικοί τελεστές: +, -, *, /, ^, ΑΚΕΡΑΙΟΙ τελεστές: div, mod ($x \text{ div } y$, $x \text{ mod } y$: $y \neq 0$ και $x, y > 0$) π.χ.:

$x \text{ mod } 2 = 0 ? \Leftrightarrow$ άρτιος ?

$x \text{ mod } y = 0 ? \Leftrightarrow$ ο x πολλαπλάσιο του y ?

$x \text{ mod } 10 =$ τελευταίο ψηφίο του x

$x \text{ div } 10 =$ ο x χωρίς το τελευταίο ψηφίο του

ανάλυση 2/3/4ψηφίου κλπ αριθμού στα ψηφία του

x	y
$x \text{ mod } y$	$x \text{ div } y$

$$x \text{ mod } y \in [0, y-1] \quad x \text{ div } y \in [0, x]$$

$$x \text{ div } y * y + x \text{ mod } y = x$$

x	y	x div y	x mod y
21	7	3	0
18	4	4	2
7	9	0	7

2ψηφιος π.χ. 53: $5=53 \text{ div } 10$, $3=53 \text{ mod } 10$

3ψηφιος π.χ. 538: $5=538 \text{ div } 100$, $3=538 \text{ div } 10 \text{ mod } 10$, $8=538 \text{ mod } 10$

Έλεγχος αν ο x έχει το πολύ 3 ψηφία

$$x \text{ div } 1000 = 0$$

Τιμή του ψηφίου των χιλιάδων

$$x \text{ div } 1000 \text{ mod } 10$$

Έλεγχος αν ο x έχει τουλάχιστον 3 ψηφία

$$x \text{ div } 100 <> 0$$

Τιμή του ψηφίου των εκατοντάδων

$$x \text{ div } 100 \text{ mod } 10$$

(§7.6) Συναρτήσεις:

$$x \text{ div } y = A_M(x/y)$$

$$\text{δεκαδικό μέρος του } X = X - A_M(X)$$

$$T_P(X) = X^{(1/2)} \quad A_M(x) \leq x < A_M(x) + 1$$

$$A_T(X) = T_P(X^2)$$

Οι τριγωνομετρικές συναρτήσεις: σε μοίρες. Π.χ. $HM(30) = 0.5$

ο ακέραιος x ισούται με το τετράγωνο άλλου ακέραιου.

$$\text{π.χ. } x = 49 = 7^2 \Leftrightarrow A_M(T_P(x)) = T_P(x)$$

1^ο δεκαδικό ψηφίο $x \in \mathbb{R}$

$$A_M(x*10) \bmod 10$$


Στρογγυλοποίηση $x \in \mathbb{R}$ προς τον πλησιέστερο ακέραιο

$$A_M(x + 0.5)$$

(§7.7) Αριθμητικές εκφράσεις: ορισμός, όλες οι μεταβλητές, πρέπει να έχουν τιμή!, ιεραρχία:

1. Παρενθέσεις, συναρτήσεις
2. Ύψωση σε δύναμη
3. Πολλαπλασιασμός και διαίρεση, div, mod
4. Πρόσθεση και αφαίρεση

$$\text{π.χ. } 2+3*4 = 14 \text{ ενώ } (2+3)*4 = 20$$



HM(X)	Υπολογισμός ημιτόνου
ΣΥΝ(X)	Υπολογισμός συνημιτόνου
ΕΦ(X)	Υπολογισμός εφαπτομένης
T_P(X)	Υπολογισμός τετραγωνικής ρίζας
ΛΟΓ(X)	Υπολογισμός φυσικού λογαρίθμου
E(X)	Υπολογισμός του e^x
A_M(X)	Ακέραιο μέρος του X
A_T(X)	Απόλυτη τιμή του X

(§7.8) Εντολή εκχώρησης: λειτουργία, σύνταξη:

<μεταβλητή> ← <έκφραση>

Κανόνες:

- το αριστερό και το δεξιό μέρος πρέπει να είναι του ίδιου τύπου
- στο αριστερό μέρος: μόνο μεταβλητή
- οι μεταβλητές στο δεξιό μέρος πρέπει να ΜΗΝ είναι απροσδιόριστες
- στο αριστερό και το δεξιό μέρος μπορεί να εμφανίζεται η ίδια μεταβλητή.

Μεταβλητές

Ακέραιες: x, y, z, m

Πραγματικές: f

Χαρακτήρες: τίτλος

Λογικές: ισχύει

Εντολή	x	y	z	m	f	τίτλος	ισχύει
	-	-	-	-	-	-	-
x ← 5	5	-	-	-	-	-	-
x ← 7	7	-	-	-	-	-	-
y ← x	7	7	-	-	-	-	-
x ← x + 3	10	7	-	-	-	-	-
z ← x + y	10	7	17	-	-	-	-
y ← x + m	10	?	17	-	-	-	-
y ← x ²	10	100	17	-	-	-	-
m ← x+2*z	10	100	17	44	-	-	-
f ← (x+z)/2	10	100	17	44	13.5	-	-
τίτλος ← x	10	100	17	44	13.5	-	-
τίτλος ← '3ο'	10	100	17	44	13.5	3ο	-
x+y ← 9	10	100	17	44	13.5	3ο	-
ισχύει ← Αληθής	10	100	17	44	13.5	3ο	Αληθής
ισχύει ← x > y	10	100	17	44	13.5	3ο	Ψευδής

στο αριστερό και το δεξιό μέρος μπορεί να εμφανίζεται η ίδια μεταβλητή

ΛΑΘΟΣ: οι μεταβλητές στο δεξιό μέρος πρέπει να ΜΗΝ είναι απροσδιόριστες

ΛΑΘΟΣ: το αριστερό και το δεξιό μέρος πρέπει να είναι του ίδιου τύπου


ΛΑΘΟΣ: στο αριστερό μέρος: μόνο μεταβλητή

(§7.9) Εντολή εισόδου ΔΙΑΒΑΣΕ:

Σύνταξη: Διάβασε <λίστα μεταβλητών>

Λειτουργία: προκαλείται μία «παύση» στην εκτέλεση των εντολών και το πρόγραμμα περιμένει από τον χρήστη την εισαγωγή τόσων τιμών, όσες και οι μεταβλητές της λίστας. Η κάθε εισαγόμενη τιμή, αποθηκεύεται στην αντίστοιχη μεταβλητή.

Παραδείγματα:



Διάβασε x
Διάβασε y
↔
Διάβασε x, y

(άνθρωπος → ΗΥ) για εισαγωγή δεδομένων (στην προστακτική, διότι ο προγραμματιστής-σκηνοθέτης διατάζει τον ΗΥ, ενώ ο χρήστης είναι ο θεατής του έργου).

Διαφορά της εντολής Διάβασε από την εκχώρηση (←)
π.χ. 2βάθμια



(§7.9) Εντολή εξόδου ΓΡΑΨΕ:

Σύνταξη: Γράψε <λίστα εκφράσεων>

Λειτουργία: το πρόγραμμα υπολογίζει τις τελικές τιμές των εκφράσεων και τις εμφανίζει στην οθόνη

Παραδείγματα:

Εντολή	Οθόνη
Γράψε 7	7
Γράψε 7^2	49
Γράψε '7^2'	7^2
$x \leftarrow 5$	
Γράψε x	5
$y \leftarrow 7$	
Γράψε x, y	5 7
Γράψε x	5
Γράψε y	7
Γράψε z	?
$\mu \leftarrow (x+y)/2$	
Γράψε 'μέσος όρος: ', μ	μέσος όρος: 6
Γράψε $x < y$	Αληθής

οι μεταβλητές πρέπει να ΜΗΝ είναι αππροσδιόριστες

(ΗΥ → άνθρωπος) για εξαγωγή αποτελεσμάτων



(§7.10) Δομή προγράμματος / Αλγορίθμου:

Πρόγραμμα <όνομα προγράμματος>
[Δήλωση συμβολικών σταθερών]
<Δήλωση μεταβλητών>

Αρχή

Αλγόριθμος <όνομα αλγορίθμου>
[Αρχή]

Σώμα εντολών:

Σώμα εντολών:

Διάβασε ...

Εκχωρήσεις ...

Γράψε ...

Διάβασε ... ή Δεδομένα // ... //

Εκχωρήσεις ...

Γράψε ... ή Εμφάνισε ή Αποτελέσματα //
... //

Τέλος

Τέλος_Προγράμματος

ΕΚΤΕΛΕΣΤΕΕΣ ΕΝΤΟΛΕΣ: υποδηλώνουν ενέργειες (π.χ. Διάβασε)

Δηλωτικές εντολές: δηλώνουν περιοχές (π.χ. Μεταβλητές)

Δεσμευμένες λέξεις: εντολές-λέξεις της γλώσσας (όλα τα παραπάνω)

(§7.10) Δομή προγράμματος / Αλγόριθμος: (π.χ. Πρόγραμμα που διαβάζει τις διαστάσεις ενός τραpezίου και εμφανίζει το εμβαδόν του)

Μη φιλικά προς τον χρήστη

Φιλικά προς τον χρήστη

Πρόγραμμα Τραπεζίο

Μεταβλητές

Πραγματικές: β, Β, υ, Εμ

Αρχή

Διάβασε β, Β, υ

$Εμ \leftarrow (\beta + B) * υ / 2$

Γράψε Εμ

Τέλος_Προγράμματος

Πρόγραμμα Τραπεζίο

Μεταβλητές

Πραγματικές: β, Β, υ, Εμ

Αρχή


Γράψε 'Δώσε διαστάσεις τραpezίου'

Διάβασε β, Β, υ

$Εμ \leftarrow (\beta + B) * υ / 2$

Γράψε 'Εμβαδό: ', Εμ


Τέλος_Προγράμματος




(§2.4) Διάγραμμα ροής (Δ.Ρ.)

ΑΠΑΙΤΕΙΤΑΙ η γνώση μετατροπής μεταξύ των 3 εννοιών:


Κώδικας ↔ Δ.Ρ. ↔ Εκφώνηση



π.χ. $MO \leftarrow (x+y+z)/3$



π.χ. $ax + b = 0$



Σχόλια

Ορισμός: οτιδήποτε μπαίνει μετά από το ειδικό σύμβολο !

Χρησιμότητα: δεν επηρεάζουν τη λειτουργία του προγράμματος, αλλά το κάνουν πιο ευανάγνωστο

π.χ. μισθός ← 1.05 * μισθός ! **προσαύξηση μισθού κατά 5%**

```
A journalist asked a programmer:-  
What makes code bad?  
No comment.
```

Ο ειδικός χαρακτήρας &: συνέχεια κώδικα και στην επόμενη γραμμή.

Αλγοριθμικές Δομές:

1. Ακολουθία
2. Επιλογή
3. Επανάληψη

(§2.4.1) Παραδείγματα δομής ακολουθίας (και Δ.Ρ.)

Πραγματικοί $a, \beta \rightarrow$ απόσταση των a και $\beta = A_T(a-\beta)$

π.χ. απόκλιση προφορικού και γραπτού βαθμού

```
Διάβασε n, γ  
ap ← A_T(n - γ)  
Γράψε ap
```

Θερμοκρασία C
 \rightarrow Θερμοκρασία
F: $(F-32)/9 = C/5$

```
Διάβασε C  
F ← 9 * C / 5 + 32  
Γράψε F
```

Κεφάλαιο, % επιτόκιο
 \rightarrow τελικό κεφάλαιο
μετά από τρία έτη

```
Διάβασε K, Ep  
K ← K + Ep/100*K  
K ← K + Ep/100*K  
K ← K + Ep/100*K  
Γράψε K
```

R κύκλου → διάμετρος,
περιφέρεια, εμβαδό,
όγκος σφαίρας με αυτήν
την ακτίνα ($4/3\pi R^3$)

Διάβασε R
 $\Delta \leftarrow 2 * R$
 $\Pi \leftarrow 2 * 3.14 * R$
 $Em \leftarrow 3.14 * R^2$
 $V \leftarrow 4/3 * 3.14 * R^3$
Γράψε Δ, Π, Em, V

α, β ορθογώνιου
παρ/γμου →
περίμετρος,
εμβαδό,
διαγώνιος

Διάβασε α, β
 $\Pi \leftarrow 2 * (α + β)$
 $Em \leftarrow α * β$
 $\Delta \leftarrow \sqrt{α^2 + β^2}$
Γράψε Π, Em, Δ

S : εμβαδό κύκλου →
εμβαδό του
περιγεγραμμένου
4γώνου

Διάβασε S
 $R \leftarrow \sqrt{S / 3.14}$
 $α \leftarrow 2 * R$
 $Em \leftarrow α^2$
Γράψε Em

(x1, y1), (x2, y2) → απόσταση των 2 σημείων

Διάβασε x1, y1, x2, y2
 $d \leftarrow \sqrt{(x1 - x2)^2 + (y1 - y2)^2}$
Γράψε d

Άθροισμα ψηφίων 3ψήφιου
θετικού ακεραίου

Διάβασε x
 $x1 \leftarrow x \text{ div } 100$
 $x2 \leftarrow x \text{ div } 10 \text{ mod } 10$
 $x3 \leftarrow x \text{ mod } 10$
 $S \leftarrow x1 + x2 + x3$
Γράψε S

Ένα παιχνίδι παίζεται κυκλικά από 5 παίκτες. Σε κάθε γύρο παίζει 1 παίκτης. Με δεδομένο τον γύρο (≥ 1) του παιχνιδιού ποιος παίκτης (1-5) παίζει;

$\text{παίκτης} \leftarrow (\text{γύρος} - 1) \text{ mod } 5 + 1$

Τι θα εμφανίσουν τα παρακάτω με τιμές εισόδου: 11, 5
Διάβασε y, x
 $z \leftarrow x \text{ mod } y$
 $y \leftarrow y \text{ div } 4$
 $x \leftarrow x \text{ div } (x + 3)$
Γράψε z, y, x

Αναγωγή χρονικού διαστήματος (sec) σε ημέρες και ότι περισσεύει σε ώρες και ότι περισσεύει σε λεπτά και ότι περισσεύει σε sec

```

Διάβασε t
d ← t div 86400
t ← t mod 86400
h ← t div 3600
t ← t mod 3600
m ← t div 60
s ← t mod 60
Γράψε d, h, m, s
    
```

"Επιμερισμός ποσότητας" π.χ. να γραφεί πρόγραμμα που διαβάζει το συνολικό ποσό επιχορήγησης τριών δήμων και τους πληθυσμούς τους. Εάν το ποσό διανέμεται αναλογικά με βάση τον πληθυσμό του κάθε δήμου, να εμφανίζεται το ποσό που θα λάβει ο κάθε δήμος

```

ΓΡΑΨΕ 'Δώσε ποσό επιχορήγησης'
ΔΙΑΒΑΣΕ poso
ΓΡΑΨΕ 'Δώσε τους πληθυσμούς των τριών δήμων'
ΔΙΑΒΑΣΕ p1 ,p2, p3
s ←-- p1+p2+p3
ΓΡΑΨΕ 'Ποσά επιχορήγησης των δήμων:'
ΓΡΑΨΕ p1/s*poso, p2/s*poso, p3/s*poso, '€'
    
```

Τριψήφιος ακέραιος x
 → x-x' (π.χ. 785 → 785 - 587)

```

Διάβασε x
x1 ← x div 100
x2 ← x div 10 mod 10
x3 ← x mod 10
sx ← x3 *100 + x2 * 10 + x1
d ← x - sx
Γράψε d
    
```

Αλυσίδα Super Market προσφέρει δωροεπιταγή 3€ για κάθε 100 πόντους αγορών. Εισαγωγή των πόντων ενός πελάτη και εμφάνιση: πόσες δωροεπιταγές δικαιούται και πόσοι πόντοι αγορών του περισσεύουν.

```

Διάβασε π
Γράψε π div 100, ' δωροεπιταγές '
Γράψε π mod 100, ' πόντοι περισσεύουν'
    
```

Συμπληρώστε το κενό ώστε να εμφανισθεί η τιμή 24

```

x ← 4
y ← x * 3
x ← x - ...
y ← x * y
Γράψε y
    
```

x	y
4	-
4	12
4-ω	12
4-ω	12(4-ω)

$$12(4-\omega) = 24 \Rightarrow \omega = 2$$

Γράψτε την ισοδύναμη έκφραση: $\frac{7}{(y-9)^{\frac{3}{4 \cdot y}}}$

$$7/(y-9)^{(3/(4 \cdot y))}$$

(§2.4.1) Παραδείγματα δομής ακολουθίας (και Δ.Ρ.)

“Ομαδοποίηση” εντολών

Αντικαταστήστε τα παρακάτω μπλοκ εντολών με μία ισοδύναμη εντολή

$$\begin{aligned}x &\leftarrow x + 1 \\x &\leftarrow x + 2 \\x &\leftarrow x + 3 \\x &\leftarrow x + 4\end{aligned}$$
 \Leftrightarrow $x \leftarrow x + 10$
$$\begin{aligned}x &\leftarrow x - 1 \\x &\leftarrow x + 4 \\x &\leftarrow x - 2 \\x &\leftarrow x + 6\end{aligned}$$
 \Leftrightarrow $x \leftarrow x + 7$
$$\begin{aligned}x &\leftarrow x * 2 \\x &\leftarrow x * 3 \\x &\leftarrow x * 4 \\x &\leftarrow x * 5\end{aligned}$$
 \Leftrightarrow $x \leftarrow x * 120$
$$\begin{aligned}x &\leftarrow x * 4 \\x &\leftarrow x / 2 \\x &\leftarrow x * 8 \\x &\leftarrow x / 4\end{aligned}$$
 \Leftrightarrow $x \leftarrow x * 4$

Μετατροπή φράσεων σε εντολές εκχώρησης

Να μετατρέψετε σε **μία** εντολή εκχώρησης κάθε μία από τις παρακάτω φράσεις:

1. Εκχώρησε στο I τον μέσο όρο των A, B, Γ
2. Αύξησε την τιμή του M κατά 2
3. Διπλασίασε την τιμή του Λ
4. Μείωσε την τιμή του X κατά την τιμή του Ψ
5. Εκχώρησε στο A το υπόλοιπο της ακέραιας διαίρεσης του A με το B.
6. Εκχώρησε στο X την τιμή Αληθής εάν $a > \beta$ διαφορετικά, εκχώρησέ της την τιμή Ψευδής
7. Εκχώρησε στο X την τιμή Αληθής/ Ψευδής εάν η λεκτική μεταβλητή o1 προηγείται αλφαβητικά της o2
8. Αύξησε το X κατά 2.3%
9. Ύψωσε το λ στην 9^η
10. Εκχώρησε στο ψ το τελευταίο ψηφίο του ακέραιου x
11. Εκχώρησε στο δ το 1^ο δεκαδικό ψηφίο του πραγματικού x
12. Εκχώρησε στο X την τιμή Αληθής/ Ψευδής εάν η ακέραια μεταβλητή λ διαιρεί ακριβώς την ακέραια μεταβλητή κ
13. Εκχώρησε στο X Αληθής/Ψευδής εάν η ακέραια μεταβλητή κ είναι πολλαπλάσιο της ακέραιας μεταβλητής λ
14. Εκχώρησε στο δ το δεκαδικό μέρος του πραγματικού x
15. Εκχώρησε στο λ την τιμή Αληθής/Ψευδής εάν η πραγματική μεταβλητή x είναι στρογγυλός ακέραιος

Ποσοστά

(αρχική τιμή, %μεταβολή_ποσ) \rightarrow τελική τιμή
τελική_τιμή \leftarrow αρχική_τιμή + αρχική_τιμή * μεταβολή_ποσ/100

(αρχική τιμή, μεταβολή) \rightarrow %μεταβολή_ποσ
μεταβολή_ποσ \leftarrow μεταβολή / αρχική_τιμή * 100

(σύνολο, υποσύνολο) \rightarrow %ποσοστό
ποσοστό \leftarrow υποσύνολο / σύνολο * 100

(σύνολο, %ποσοστό) \rightarrow υποσύνολο
υποσύνολο \leftarrow σύνολο * ποσοστό/100

Αντιμετάθεση τιμών δύο μεταβλητών x και y

$x \leftarrow y$
 $y \leftarrow x$


	x	y
	5	7
$x \leftarrow y$	7	7
$y \leftarrow x$	7	7

ΛΑΘΟΣ

$tmp \leftarrow x$
 $x \leftarrow y$
 $y \leftarrow tmp$

	x	y	tmp
	5	7	.
$tmp \leftarrow x$	5	7	5
$x \leftarrow y$	7	7	5
$y \leftarrow tmp$	7	5	5

ΣΩΣΤΟ




ΕΝΑΛΛΑΚΤΙΚΑ (μόνο για x, y: αριθμητικές)

$x \leftarrow x + y$
 $y \leftarrow x - y$
 $x \leftarrow x - y$

$tmp \leftarrow x + y$
 $y \leftarrow tmp - y$
 $x \leftarrow tmp - y$

Κυκλική αντιμετάθεση τιμών τριών μεταβλητών x, y, z



```
tmp ← x  
x ← y  
y ← z  
z ← tmp
```

x	y	z	tmp
5	7	9	-
5	7	9	5
7	7	9	5
7	9	9	5
7	9	5	5

Εισαγωγή 4ψήφιου θετικού ακέραιου x και 1ψήφιου y → 5ψήφιος με τον y: μεσαίο ψηφίο

Εισαγωγή ενός έτους (>0) και εμφάνιση της δεκαετίας και του αιώνα του. Π.χ.
έτος = 1987 → 9η δεκαετία του 20ου αιώνα
έτος = 2003 → 1η δεκαετία του 21ου αιώνα

```
Διάβασε x, y  
x1 ← x div 100  
x2 ← x mod 100  
z ← x1 * 1000 + y * 100 + x2  
Γράψε z
```

```
Διάβασε έτος  
δεκαετία <-- έτος div 10 mod 10 + 1  
αιώνας <-- έτος div 100 + 1  
Γράψε δεκαετία, 'η δεκαετία του ', αιώνας, 'ου αιώνα'
```

(§2.4.2-4 - §8.1) Δομή επιλογής

α. Λογική έκφραση: <έκφραση1> συγκριτικός τελεστής <έκφραση 2>

β. Τελεστές σύγκρισης

Συγκριτικοί τελεστές		
Τελεστής	Ελεγχόμενη σχέση	Παράδειγμα
=	Ισότητα	Αριθμός=0
<>	Ανισότητα	Όνομα1 <> 'Κώστας'
>	Μεγαλύτερο από	Τιμή>10000
>=	Μεγαλύτερο ή ίσο	$X+Y \geq (A+B)/\Gamma$
<	Μικρότερο από	$B^2-4*A*\Gamma < 0$
<=	Μικρότερο ή ίσο	Βάρος <= 500

Εκφράσεις σύγκρισης του x:

- "το πολύ N"/"έως και N" ($x \leq N$),
- "λιγότερο από N"/"κάτω του N" ($x < N$),
- "περισσότερο από N"/"άνω του N" ($x > N$),
- "τουλάχιστον N"/"από N και πάνω" ($x \geq N$),
- "από N έως και M" ($x \geq N$ ΚΑΙ $x \leq M$),
- "άνω του N και κάτω του M"/"μεταξύ N και M" ($x > N$ ΚΑΙ $x < M$)

Παρατηρήσεις:

α) η σύγκριση των λεκτικών τιμών γίνεται με αλφαβητική σειρά.
π.χ. 'ΑΛΕΞΑΝΔΡΟΣ' > 'ΑΛΕΞΑΝΔΡΑ', 'ΜΕΓΑΛΟΣ' < 'ΜΙΚΡΟΣ'

β) η σύγκριση λογικών δεδομένων έχει έννοια μόνο στην περίπτωση του ίσου (=) και του διάφορου (<>)

γ) η μεταβλητή τύπου χαρακτήρων λέξη ξεκινάει με Κ:
(λέξη >= 'Κ' ΚΑΙ λέξη < 'Λ')

(§2.4.2-4 - §8.1) Δομή επιλογής

γ. Σύνθετη λογική έκφραση:

<λογική έκφραση 1> λογικός τελεστής <λογική έκφραση 2>

δ. Λογικοί τελεστές

i. Τελεστής ΚΑΙ (σύζευξη)

Σύνταξη: <λογική έκφραση 1> ΚΑΙ <λογική έκφραση 2>

Λειτουργία: ισούται με Αληθής όταν όλες οι εκφράσεις που συνδέει, είναι Αληθείς

Παράδειγμα: $x > 0$ ΚΑΙ $x \leq 12$ ($\Leftrightarrow x \in (0, 12]$)

ii. Τελεστής Η (διάζευξη)

Σύνταξη: <λογική έκφραση 1> Η <λογική έκφραση 2>

Λειτουργία: ισούται με Αληθής όταν τουλάχιστον μία από τις εκφράσεις που συνδέει, είναι Αληθής

Παράδειγμα: $x \leq 0$ Η $x > 12$ ($\Leftrightarrow x \in (-\infty, 0] \cup (12, +\infty)$)

iii. Τελεστής ΟΧΙ (άρνηση)

Σύνταξη: ΟΧΙ <λογική έκφραση>

Λειτουργία: Ισούται με την αντίθετη τιμή της έκφρασης

Παραδείγματα:

1. ΟΧΙ ($x = 7$) ($\Leftrightarrow x \neq 7$)
2. ΟΧΙ ($y > 11$) ($\Leftrightarrow y \leq 11$)
3. ΟΧΙ ($z > -3$ ΚΑΙ $z \leq 18$) ($\Leftrightarrow z \leq -3$ Η $z > 18$)
4. ΟΧΙ ($\omega \leq 0$ Η $\omega > 21$) ($\Leftrightarrow \omega > 0$ ΚΑΙ $\omega \leq 21$)

- $x \in [0, 1) \Leftrightarrow$
- $x \geq 0$ ΚΑΙ $x < 1$
 - ΟΧΙ($x < 0$ Η $x \geq 1$)
 - ΟΧΙ($x < 0$) ΚΑΙ $x < 1$
 - $x \geq 0$ ΚΑΙ ΟΧΙ($x \geq 1$)
 - ΟΧΙ($x < 0$) ΚΑΙ ΟΧΙ($x \geq 1$)

Πίνακας Αληθείας


Πρόταση Α	Πρόταση Β	Α ή Β	Α και Β	όχι Α
Αληθής	Αληθής	Αληθής	Αληθής	Ψευδής
Αληθής	Ψευδής	Αληθής	Ψευδής	Ψευδής
Ψευδής	Αληθής	Αληθής	Ψευδής	Αληθής
Ψευδής	Ψευδής	Ψευδής	Ψευδής	Αληθής

ε. Ιεραρχία λογικών τελεστών:

1. Παρενθέσεις
2. ΟΧΙ
3. ΚΑΙ
4. Η

στ. Ιεραρχία τελεστών:

1. Αριθμητικοί (+, -, κλπ.)
2. Συγκριτικοί (>, =, κλπ.)
3. Λογικοί (ΟΧΙ, ΚΑΙ, Η)



(§2.4.2-4 - §8.1) Δομή επιλογής

ζ. Απλή Δομή Αν – ΤέλοςΑν Σύνταξη:

Αν <συνθήκη> τότε
<εντολές>

ΤέλοςΑν

ή Αν <συνθήκη> τότε <εντολή>

Πρόγραμμα που διαβάζει: τις ημέρες ενοικίασης ενός αυτοκινήτου, το ημερήσιο κόστος ενοικίασης καθώς και αν θα έχει μικτή ασφάλιση (ΝΑΙ/ΟΧΙ). Να εμφανίζει το συνολικό κόστος ενοικίασης εάν η μικτή ασφάλιση κοστίζει επιπλέον 5€ ημερησίως.

ΠΡΟΓΡΑΜΜΑ Αν_TelosAn

ΜΕΤΑΒΛΗΤΕΣ

ΑΚΕΡΑΙΕΣ: hmeres

ΠΡΑΓΜΑΤΙΚΕΣ: kostosHm, k

ΧΑΡΑΚΤΗΡΕΣ: asf

ΑΡΧΗ

ΔΙΑΒΑΣΕ hmeres, kostosHm, asf

k ←-- kostosHm * hmeres



ΑΝ asf = 'ΝΑΙ' ΤΟΤΕ

k ←-- k + 5*hmeres

ΤΕΛΟΣ_ΑΝ

ΓΡΑΨΕ k

ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ



(§2.4.2-4 - §8.1) Δομή επιλογής

η. Σύνθετη Δομή Αν - Αλλιώς - ΤέλοςΑν

Σύνταξη:


Αν <συνθήκη> τότε

<εντολές1>

Αλλιώς

<εντολές2>

ΤέλοςΑν



π.χ. θετικός ακέραιος $a \rightarrow$ άρτιος / περιττός

Διάβασε a


Αν $(a \bmod 2 = 0)$ τότε

Γράψε 'άρτιος'

Αλλιώς

Γράψε 'περιττός'

ΤέλοςΑν





(§2.4.2-4 - §8.1) Δομή επιλογής

θ. Εμφωλευμένες Αν ΚΑΝΟΝΑΣ ΕΜΦΩΛΕΥΣΗΣ: Κάθε Αλλιώς και ΤέλοςΑν αντιστοιχεί στο πλησιέστερο ανοικτό Αν

Να διαβάζονται δύο αριθμοί που αντιστοιχούν στο ύψος και βάρος ενός άνδρα. Να εκτυπώνεται ότι ο άνδρας είναι “ελαφρύς”, αν το βάρος του είναι κάτω από 80 κιλά, ή να εκτυπώνεται “βαρύς” στην αντίθετη περίπτωση. Επίσης να εκτυπώνεται “κοντός” αν το ύψος του είναι κάτω από 1.70, αλλιώς να εκτυπώνεται “ψηλός”.


```
1  ΠΡΟΓΡΑΜΜΑ baros_ypsos
2  ΜΕΤΑΒΛΗΤΕΣ
3  ΠΡΑΓΜΑΤΙΚΕΣ: b, y
4  ΑΡΧΗ
5  ΔΙΑΒΑΣΕ b, y
6  ΑΝ b < 80 ΤΟΤΕ
7  ΑΝ y < 1.70 ΤΟΤΕ
8  ΓΡΑΨΕ 'ελαφρύς-κοντός'
9  ΑΛΛΙΩΣ
10 ΓΡΑΨΕ 'ελαφρύς-ψηλός'
11 ΤΕΛΟΣ_ΑΝ
12 ΑΛΛΙΩΣ
13 ΑΝ y < 1.70 ΤΟΤΕ
14 ΓΡΑΨΕ 'βαρύς-κοντός'
15 ΑΛΛΙΩΣ
16 ΓΡΑΨΕ 'βαρύς-ψηλός'
17 ΤΕΛΟΣ_ΑΝ
18 ΤΕΛΟΣ_ΑΝ
19 ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ
```



(§2.4.2-4 - §8.1) Δομή επιλογής

θ. Εμφωλευμένες Αν

Εισαγωγή των συντεταγμένων (x,y) ενός σημείου $(x,y \neq 0)$ και εμφάνιση του τεταρτημορίου (1-4) στο οποίο βρίσκεται:



```
ΠΡΟΓΡΑΜΜΑ tetarthmoria
ΜΕΤΑΒΛΗΤΕΣ
  ΠΡΑΓΜΑΤΙΚΕΣ: x, y
ΑΡΧΗ
  ΔΙΑΒΑΣΕ x, y
  ΑΝ x >0 ΤΟΤΕ
    ΑΝ y >0 ΤΟΤΕ
      ΓΡΑΨΕ 1
    ΑΛΛΙΩΣ
      ΓΡΑΨΕ 2
    ΤΕΛΟΣ_ΑΝ
  ΑΛΛΙΩΣ
    ΑΝ y >0 ΤΟΤΕ
      ΓΡΑΨΕ 4
    ΑΛΛΙΩΣ
      ΓΡΑΨΕ 3
    ΤΕΛΟΣ_ΑΝ
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ
```

ή

```
ΠΡΟΓΡΑΜΜΑ tetarthmoria2
ΜΕΤΑΒΛΗΤΕΣ
  ΠΡΑΓΜΑΤΙΚΕΣ: x, y
ΑΡΧΗ
  ΔΙΑΒΑΣΕ x, y
  ΑΝ x>0 ΚΑΙ y>0 ΤΟΤΕ
    ΓΡΑΨΕ 1
  ΑΛΛΙΩΣ_ΑΝ x>0 ΚΑΙ y<0 ΤΟΤΕ
    ΓΡΑΨΕ 2
  ΑΛΛΙΩΣ_ΑΝ x<0 ΚΑΙ y<0 ΤΟΤΕ
    ΓΡΑΨΕ 3
  ΑΛΛΙΩΣ
    ΓΡΑΨΕ 4
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ
```

ή



```
ΠΡΟΓΡΑΜΜΑ tetarthmoria3
ΜΕΤΑΒΛΗΤΕΣ
  ΠΡΑΓΜΑΤΙΚΕΣ: x, y
ΑΡΧΗ
  ΔΙΑΒΑΣΕ x, y
  ΑΝ x>0 ΚΑΙ y>0 ΤΟΤΕ
    ΓΡΑΨΕ 1
  ΤΕΛΟΣ_ΑΝ
  ΑΝ x>0 ΚΑΙ y<0 ΤΟΤΕ
    ΓΡΑΨΕ 2
  ΤΕΛΟΣ_ΑΝ
  ΑΝ x<0 ΚΑΙ y<0 ΤΟΤΕ
    ΓΡΑΨΕ 3
  ΤΕΛΟΣ_ΑΝ
  ΑΝ x<0 ΚΑΙ y>0 ΤΟΤΕ
    ΓΡΑΨΕ 4
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ
```

(§2.4.2-4 - §8.1) Δομή επιλογής

1. Δομή Αν πολλαπλής επιλογής

Αν $\Sigma 1$ τότε
 εντολές1
Αλλιώς Αν $\Sigma 2$ τότε
 εντολές2
...
Αλλιώς Αν Σv τότε
 εντολέςv
[Αλλιώς
 εντολέςv+1]
Τέλος Αν


Εκτελούνται οι εντολές που βρίσκονται στο αντίστοιχο τμήμα, όταν η συνθήκη είναι αληθής. Η εκτέλεση του προγράμματος συνεχίζεται με την εντολή που ακολουθεί τη δήλωση ΤΕΛΟΣ_ΑΝ



(§2.4.2-4 - §8.1) Δομή επιλογής

1. Δομή Αν πολλαπλής επιλογής

π.χ. δίν. βαθμός $\beta \in [0-20] \rightarrow$ χαρακτηρισμός :
κακός (≤ 9), μέτριος (9, 14], καλός (14, 16],
πολύ καλός (16, 18], άριστος (18, 20]



Διάβασε β

Αν ($\beta \leq 9$) τότε

Γράψε 'κακός'

Αλλιώς Αν ($\beta \leq 14$) τότε

Γράψε 'μέτριος'

Αλλιώς Αν ($\beta \leq 16$) τότε

Γράψε 'καλός'


Αλλιώς Αν ($\beta \leq 18$) τότε

Γράψε 'πολύ καλός'

Αλλιώς

Γράψε 'άριστος'

Τέλος Αν



κ. Δομή Επίλεξε

```
ΕΠΙΛΕΞΕ <έκφραση>
  ΠΕΡΙΠΤΩΣΗ <λίστα_τιμών_1>
    <εντολές_1>
  ΠΕΡΙΠΤΩΣΗ <λίστα_τιμών_2>
    <εντολές_2>
  .....
  ΠΕΡΙΠΤΩΣΗ ΑΛΛΙΩΣ
    <εντολές_αλλιώς>
ΤΕΛΟΣ_ΕΠΙΛΟΓΩΝ
```

Παράδειγμα 1 – Κωδικός καταστημάτων

Να αναπτύξετε πρόγραμμα σε ΓΛΩΣΣΑ που να διαβάσει τον κωδικό ενός καταστήματος και να εμφανίζει την πόλη στην οποία ανήκει. Τα καταστήματα της Αθήνας έχουν τους κωδικούς 1, 2, 3, 4 και τα καταστήματα της Θεσσαλονίκης έχουν τους κωδικούς 5 και 6. Αν δώσετε κάποιον άλλον αριθμό, να εμφανίζεται το μήνυμα «Δεν υπάρχει αυτός ο κωδικός καταστήματος».

```
ΓΡΑΨΕ ' Δώσε τον κωδικό του καταστήματος: '
ΔΙΑΒΑΣΕ κωδικός
ΕΠΙΛΕΞΕ κωδικός
  ΠΕΡΙΠΤΩΣΗ 1, 2, 3, 4
    ΓΡΑΨΕ 'Αθήνα'
  ΠΕΡΙΠΤΩΣΗ 5, 6
    ΓΡΑΨΕ 'Θεσσαλονίκη'
  ΠΕΡΙΠΤΩΣΗ ΑΛΛΙΩΣ
    ΓΡΑΨΕ ' Δεν υπάρχει αυτός ο κωδικός καταστήματος.
ΤΕΛΟΣ_ΕΠΙΛΟΓΩΝ
```

<έκφραση> :

είναι μια μεταβλητή, η τιμή της οποίας θα ελεγχθεί με τις τιμές που δίνονται στις ΠΕΡΙΠΤΩΣΕΙΣ και ανάλογα σε ποια ΠΕΡΙΠΤΩΣΗ ανήκει θα εκτελεστούν οι αντίστοιχες εντολές ή η πράξη, που υπολογίζει την τιμή της.

Δηλαδή, η <έκφραση> μπορεί να είναι:

- Μεταβλητή
- Αριθμητική πράξη
- Συγκριτική πράξη

<λίστα_τιμών_N>:

οι τιμές που μπορεί να πάρει μια έκφραση. Οι τιμές αυτές μπορεί να είναι διακριτές τιμές, περιοχή τιμών από...έως ή να υπακούν σε μια συνθήκη.

Κατά την εκτέλεση της εντολής υπολογίζεται η τιμή της έκφρασης και στη συνέχεια εκτελούνται οι εντολές που ανήκουν στην αντίστοιχη περίπτωση τιμών. Στην περίπτωση που η τιμή έκφρασης δεν αντιστοιχεί σε καμία περίπτωση, τότε εκτελούνται οι εντολές της ΠΕΡΙΠΤΩΣΗΣ_ΑΛΛΙΩΣ. Η ΠΕΡΙΠΤΩΣΗΣ_ΑΛΛΙΩΣ είναι προαιρετική.

Η εκτέλεση του προγράμματος συνεχίζεται με την εντολή που ακολουθεί μετά το ΤΕΛΟΣ_ΕΠΙΛΟΓΩΝ.

⊕: συμπαγής δομή

κ. Δομή Επίλεξε

Παράδειγμα 2 – Άρτιος ή περιττός αριθμός
Να αναπτύξετε πρόγραμμα σε ΓΛΩΣΣΑ που να διαβάζει έναν ακέραιο αριθμό και στη συνέχεια να τυπώνει αν ο αριθμός είναι άρτιος ή περιττός.

```
ΔΙΑΒΑΣΕ Χ
ΕΠΙΛΕΞΕ Χ MOD 2
    ΠΕΡΙΠΤΩΣΗ 0
        ΓΡΑΨΕ 'Άρτιος'
    ΠΕΡΙΠΤΩΣΗ ΑΛΛΙΩΣ
        ΓΡΑΨΕ 'Περιττός'
ΤΕΛΟΣ_ΕΠΙΛΟΓΩΝ
```

Εισαγωγή θετικού ακέραιου
και έλεγχος αν είναι:
μονοψήφιος/διψήφιος/άλλος

Διάβασε x
Επίλεξε x
περίπτωση από 0 έως 9 ! ή 0..9
Γράψε 'μονοψήφιος'
περίπτωση από 10 έως 99 ! ή 10..99
Γράψε 'διψήφιος'
περίπτωση Αλλιώς
Γράψε 'άλλος'
Τέλος Επιλογών

Παράδειγμα 3 – Τέλη κυκλοφορίας αυτοκινήτων
Η εφορία κάθε τέλος του έτους φορολογεί τα αυτοκίνητα ανάλογα με τον κυβισμό τους, σύμφωνα με τον παρακάτω πίνακα. Να αναπτύξετε πρόγραμμα σε ΓΛΩΣΣΑ που να διαβάζει τον κυβισμό του αυτοκινήτου, να υπολογίζει τον φόρο που του αναλογεί και να τυπώνει το αντίστοιχο ποσό.

Κυβισμός	Φόρος
0 έως 1000	100€
1001 έως 1299	120€
1300 έως 1800	250€
1801 και άνω	600€

```
ΔΙΑΒΑΣΕ Κ
ΑΝ Κ <= 1000 ΤΟΤΕ
    ΓΡΑΨΕ 'ΤΕΛΗ = 100€'
ΑΛΛΙΩΣ_ΑΝ Κ <= 1299 ΤΟΤΕ
    ΓΡΑΨΕ 'ΤΕΛΗ = 120€'
ΑΛΛΙΩΣ_ΑΝ Κ <= 1800 ΤΟΤΕ
    ΓΡΑΨΕ 'ΤΕΛΗ = 250€'
ΑΛΛΙΩΣ
    ΓΡΑΨΕ 'ΤΕΛΗ = 600€'
ΤΕΛΟΣ_ΑΝ
```



```
ΔΙΑΒΑΣΕ Κ
ΕΠΙΛΕΞΕ Κ
    ΠΕΡΙΠΤΩΣΗ <= 1000
        ΓΡΑΨΕ 'ΤΕΛΗ = 100€'
    ΠΕΡΙΠΤΩΣΗ <= 1299
        ΓΡΑΨΕ 'ΤΕΛΗ = 120€'
    ΠΕΡΙΠΤΩΣΗ <= 1800
        ΓΡΑΨΕ 'ΤΕΛΗ = 250€'
    ΠΕΡΙΠΤΩΣΗ ΑΛΛΙΩΣ
        ΓΡΑΨΕ 'ΤΕΛΗ = 600€'
ΤΕΛΟΣ_ΕΠΙΛΟΓΩΝ
```

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M1. Εύρεση min/max 3,4... ποσοτήτων (διαδοχικά Αν-ΤέλοςΑν). Για 2 ποσότητες, Αν-Αλλιώς-ΤέλοςΑν

Γενικά: εύρεση π.χ. $\max \{a_1, a_2, \dots, a_n\}$

$\max \leftarrow a_1$

Αν ($a_2 > \max$) τότε

$\max \leftarrow a_2$

ΤέλοςΑν

Αν ($a_3 > \max$) τότε

$\max \leftarrow a_3$

ΤέλοςΑν


...

Αν ($a_n > \max$) τότε

$\max \leftarrow a_n$

ΤέλοςΑν

Γράψε \max



\max 3 τιμών:

Διάβασε x, y, z

$\max \leftarrow x$

Αν ($y > \max$) τότε

$\max \leftarrow y$

ΑλλιώςΑν ($z > \max$) τότε

$\max \leftarrow z$

ΤέλοςΑν

! Είναι σωστό;

$$\max(x, y) = (x+y)/2 + A_T(x-y)/2$$

$$\min(x, y) = (x+y)/2 - A_T(x-y)/2$$

(x, y : αριθμητικές)

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M2. Εύρεση έννοιας με min/max γνώρισμα. π.χ. 4 μάρκες αυτοκινήτων, 4 τιμές → ακριβότερη μάρκα (χωρίς ισοτιμία)

Διάβασε $\mu_1, \mu_2, \mu_3, \mu_4$

Διάβασε $\tau_1, \tau_2, \tau_3, \tau_4$

$\max \leftarrow \tau_1$

$\mu_{\max} \leftarrow \mu_1$

Αν $(\tau_2 > \max)$ τότε

$\max \leftarrow \tau_2$

$\mu_{\max} \leftarrow \mu_2$

ΤέλοςΑν

Αν $(\tau_3 > \max)$ τότε

$\max \leftarrow \tau_3$

$\mu_{\max} \leftarrow \mu_3$

ΤέλοςΑν


Αν $(\tau_4 > \max)$ τότε

$\max \leftarrow \tau_4$

$\mu_{\max} \leftarrow \mu_4$

ΤέλοςΑν

Γράψε μ_{\max}



Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M3. Ασκήσεις σύνθετων ελέγχων με χρήση των τελεστών: ΚΑΙ, Η, ΟΧΙ
π.χ. Βαθμοί ενός φοιτητή σε 3 μαθήματα → περνάει το δμηνο (έχει σε όλα τουλάχιστον τη βάση (5) ή έχει μέσο όρο άνω του 7)

Διάβασε B1, B2, B3

$MO \leftarrow (B1+B2+B3)/3$


Αν $(B1 \geq 5 \text{ ΚΑΙ } B2 \geq 5 \text{ ΚΑΙ } B3 \geq 5 \text{ Η } MO > 7)$ τότε

Γράψε 'Περνάει'

Αλλιώς

Γράψε 'Κόβεται'

Τέλος



Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M4. Κλαδικές συναρτήσεις

π.χ. $x \rightarrow f(x)$:

$$f(x) = \begin{cases} x - 1 & \text{για } x < 0 \\ x^2 - 3 \cdot x & \text{για } 0 \leq x \leq 1 \\ \frac{x+1}{x-4} & \text{για } x > 1 \end{cases}$$

Διάβασε x

Αν $(x < 0)$ τότε

$f \leftarrow x-1$

Γράψε f

Αλλιώς Αν $(x \leq 1)$ τότε

$f \leftarrow x^2 - 3 \cdot x$

Γράψε f

Αλλιώς

Αν $(x < 4)$ τότε

$f \leftarrow (x+1)/(x-4)$


Γράψε f

Αλλιώς

Γράψε 'αδύνατη'

Τέλος Αν

Τέλος Αν



Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M5. Πίνακες περιπτώσεων

π.χ. Μία εταιρεία ταχυδρομικών υπηρεσιών εφαρμόζει για τα έξοδα αποστολής ταχυδρομικών επιστολών εσωτερικού και εξωτερικού, χρέωση σύμφωνα με τον παρακάτω πίνακα:

Να γράψετε πρόγραμμα το οποίο:

- Να διαβάζει το βάρος της επιστολής.
- Να διαβάζει τον προορισμό της επιστολής. Η τιμή "ΕΣ" δηλώνει προορισμό εσωτερικού και η τιμή "ΕΞ" δηλώνει προορισμό εξωτερικού.
- Να υπολογίζει τα έξοδα αποστολής ανάλογα με τον προορισμό και το βάρος της επιστολής.
- Να εκτυπώνει τα έξοδα αποστολής.

Βάρος επιστολής σε γραμμάρια	Χρέωση εσωτερικού σε €	Χρέωση εξωτερικού σε €
από 0 έως και 500	2,0	4,8
από 501 έως και 1000	3,5	7,2
από 1001 και άνω	4,6	11,5

Διάβασε β, πρ
Αν ($\beta \leq 500$) τότε
Αν (πρ = 'ΕΣ') τότε
κ ← 2
Αλλιώς
κ ← 4.8
ΤέλοςΑν
ΑλλιώςΑν ($\beta \leq 1000$) τότε
Αν (πρ = "ΕΣ") τότε
κ ← 3.5
Αλλιώς
κ ← 7.2
ΤέλοςΑν
Αλλιώς
Αν (πρ = "ΕΣ") τότε
κ ← 4.6
Αλλιώς
κ ← 11.5
ΤέλοςΑν
ΤέλοςΑν
Γράψε κ


Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M6. Κλιμακωτή χρέωση

Π.χ. πρόγραμμα που διαβάζει το ετήσιο εισόδημα και τον αριθμό των παιδιών ενός φορολογούμενου και υπολογίζει κλιμακωτά τον φόρο του με βάση τα παρακάτω. Εάν η έκπτωση φόρου προκύψει μεγαλύτερη από τον φόρο, ο φόρος να μηδενίζεται.

Εισόδημα	Φόρος (%)
0-3000	0
3001-7000	10
7001-15000	14
15001-20000	20
20001-άνω	30

Παιδιά	Έκπτωση φόρου (€)
1	100
2	250
3	400
4 και άνω	550



Διάβασε ϵ , π Αν ($\epsilon \leq 3000$) τότε $\phi \leftarrow 0$ Αλλιώς Αν ($\epsilon \leq 7000$) τότε $\phi \leftarrow 10/100 * (\epsilon - 3000)$ Αλλιώς Αν ($\epsilon \leq 15000$) τότε $\phi \leftarrow 10/100 * 4000 + 14/100 * (\epsilon - 7000)$ Αλλιώς Αν ($\epsilon \leq 20000$) τότε $\phi \leftarrow 10/100 * 4000 + 14/100 * 8000 + 20/100 * (\epsilon - 15000)$ Αλλιώς $\phi \leftarrow 10/100 * 4000 + 14/100 * 8000 + 20/100 * 5000 + 30/100 * (\epsilon - 20000)$ Τέλος Αν	Αν ($\pi = 1$) τότε $\phi \leftarrow \phi - 100$ Αλλιώς Αν ($\pi = 2$) τότε $\phi \leftarrow \phi - 250$ Αλλιώς Αν ($\pi = 3$) τότε $\phi \leftarrow \phi - 400$ Αλλιώς Αν ($\pi \geq 4$) τότε $\phi \leftarrow \phi - 550$ Τέλος Αν Αν ($\phi < 0$) τότε $\phi \leftarrow 0$ Τέλος Αν Γράψε ϕ
--	---

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M7. Διερεύνηση πραγματικού αριθμού

Π.χ. πρόγραμμα που διαβάζει έναν θετικό πραγματικό αριθμό και ελέγχει εάν είναι:

- ακέραιος, οπότε και τον χαρακτηρίζει ως άρτιο / περιττό
- πραγματικός, οπότε και τον στρογγυλοποιεί στον πλησιέστερο ακέραιο (π.χ. $7.3 \rightarrow 7$, $7.8 \rightarrow 8$, $7.5 \rightarrow 8$)

```
Διάβασε x
Αν (x = A_M(x)) τότε
  Αν (x mod 2 = 0) τότε
    Γράψε "άρτιος"
  Αλλιώς
    Γράψε "περιττός"
  ΤέλοςΑν
```

```
Αλλιώς
  ΔM ← x - A_M(x)
  Αν (ΔM >= 0.5) τότε
    ΣX ← A_M(x) + 1
  Αλλιώς
    ΣX ← A_M(x)
  ΤέλοςΑν
  Γράψε ΣX
ΤέλοςΑν
```

```
ή
ΣX ← A_M(x+0.5)
```

M8. Μενού επιλογών

Π.χ. πρόγραμμα που με μενού επιλογών εκτελεί τις παρακάτω λειτουργίες:

1. Εμβαδό κύκλου
2. Εμβαδό τραπεζίου
3. Εμβαδό κυλίνδρου ($2\pi R^2 + 2\pi Rh$)

```
Γράψε "1. Εμβαδό κύκλου"
Γράψε "2. Εμβαδό τραπεζίου "
Γράψε "3. Εμβαδό κυλίνδρου "
Γράψε "Διάλεξε (1-3)"
```

```
Διάβασε επ
Αν (επ = 1) τότε
```

```
  Διάβασε ρ
  EM ← 3.14*ρ^2
  Γράψε EM
```

```
ΑλλιώςΑν (επ = 2) τότε
```

```
  Διάβασε β, Β, υ
  EM ← (β + Β) * υ / 2
  Γράψε EM
```

```
ΑλλιώςΑν (επ = 3) τότε
```

```
  Διάβασε R, h
  EM ← 2*3.14*R^2 + 2*3.14*R*h
  Γράψε EM
```

```
Αλλιώς
```

```
  Γράψε "Λάθος επιλογή"
ΤέλοςΑν
```

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M9. Εφαρμογή των div και mod

Έστω ότι η χρέωση ενός κινητού είναι 0.30€ ανά πεντάλεπτο ομιλίας (η χρέωση ενός πενταλέπτου γίνεται ακόμη και αν δεν έχει εξαντληθεί). π.χ. για χρόνο ομιλίας 28 λεπτά, χρεώνονται 6 πεντάλεπτα. Εισαγωγή του χρόνου ομιλίας και εμφάνιση της χρέωσης (€)

Διάβασε t

$\lambda 5 \leftarrow t \text{ div } 5$

Αν $(t \text{ mod } 5 <> 0)$ τότε

$\lambda 5 \leftarrow \lambda 5 + 1$

ΤέλοςΑν

$x \leftarrow 0.3 * \lambda 5$

Γράψε x

Εργάτης πληρώνεται προς 10€ την ώρα. Αν την τελευταία ώρα δουλέψει τουλάχιστον 30 λεπτά την πληρώνεται ολόκληρη διαφορετικά, δεν την πληρώνεται. Εισαγωγή των συνολικών λεπτών εργασίας και υπολογισμός της αμοιβής του.

Διάβασε λ

$\omega \leftarrow \lambda \text{ div } 60$

Αν $(\lambda \text{ mod } 60 \geq 30)$ τότε

$\omega \leftarrow \omega + 1$

ΤέλοςΑν

Γράψε $\omega * 10$

Γενικά: «παρτίδες» των N για σύνολο X στοιχείων

π.χ.

X: άτομα συμμετοχής σε εκδρομή

N=50 η χωρητικότητα ενός πούλμαν

π: απαραίτητος αριθμός πούλμαν

$\pi \leftarrow X \text{ div } 50$

Αν $(X \text{ mod } 50 <> 0)$ τότε

$\pi \leftarrow \pi + 1$

ΤέλοςΑν

ή

$\pi \leftarrow (X + 49) \text{ div } 50$

ή

$\pi \leftarrow (X-1) \text{ div } 50 + 1$


Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M10. Έγκυρη ημερομηνία

Να γραφεί πρόγραμμα που διαβάζει μία ημερομηνία μέσω τριών ακεραίων (d/m/y) και ελέγχει την εγκυρότητά της. Εάν δηλ. το d (ημέρες) είναι εντός του διαστήματος [1, md], όπου md = 28/29/30/31 αναλόγως των m (μήνας) και y (έτος). Υπενθυμίζεται ότι οι μήνες 4,6,9,11 έχουν 30 ημέρες. Ο Φεβρουάριος έχει 28 ημέρες για τα μη δίσεκτα έτη και 29 για τα δίσεκτα (έτη που είναι πολλαπλάσια του 4 και όχι του 100 ή είναι πολλαπλάσια του 400). Όλοι οι υπόλοιποι μήνες έχουν 31 ημέρες.

```

ΔΙΑΒΑΣΕ d, m, y
ok <-- ΑΛΗΘΗΣ
ΑΝ m < 1 Η m > 12 ΤΟΤΕ
    ok <-- ΨΕΥΔΗΣ
ΑΛΛΙΩΣ
    ΑΝ m=4 Η m=6 Η m=9 Η m=11 ΤΟΤΕ
        md <-- 30
    ΑΛΛΙΩΣ_ΑΝ m=2 ΤΟΤΕ
        ΑΝ y mod 4 = 0 ΚΑΙ y mod 100 <> 0 Η y mod 400 = 0 ΤΟΤΕ
            md <-- 29
        ΑΛΛΙΩΣ
            md <-- 28
    ΤΕΛΟΣ_ΑΝ
ΑΛΛΙΩΣ
    md <-- 31
ΤΕΛΟΣ_ΑΝ
ΑΝ d < 1 Η d > md ΤΟΤΕ
    ok <-- ΨΕΥΔΗΣ
ΤΕΛΟΣ_ΑΝ
ΓΡΑΨΕ ok
    
```



M11. Πίνακας τιμών μεταβλητών

$a \leftarrow 7$

$\beta \leftarrow a \wedge 2$

$\gamma \leftarrow 5$

Αν ($\beta \bmod 2 = 1$) **και** ($\gamma \geq 5$) **τότε**

$\gamma \leftarrow \gamma + 2$

Αν ($\gamma > a$) **τότε**

$a \leftarrow a \wedge 3$

$\gamma \leftarrow \gamma \wedge 2$

Αλλιώς

$a \leftarrow a * 4$

$\gamma \leftarrow \gamma * 2$

Τέλος_Αν

Γράψε a, β, γ

Τέλος_Αν

$a \leftarrow a \bmod \beta$

$\beta \leftarrow \beta \text{ div } \gamma$


$\gamma \leftarrow a \bmod \gamma$

Γράψε a, β, γ

α	β	γ	οθόνη
7	49	5	
7	49	7	
28	49	14	28, 49, 14
28	3	0	28, 3, 0

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M12. Μετατροπή: Αλγόριθμος \leftrightarrow ΔΡ



Πρόγραμμα ΔΡ
Μεταβλητές
Ακέραιες: X, Y
Αρχή
Διάβασε X
Αν $(X \bmod 2 = 0)$ τότε
 $Y \leftarrow X \text{ div } 2$
 Αν $(Y \leq 10)$ τότε
 $Y \leftarrow 2 * X + Y$
 ΤέλοςΑν
Αλλιώς
 $Y \leftarrow X^2$
ΤέλοςΑν
Γράψε Y
ΤέλοςΠρογράμματος

M13. Ομόσημοι αριθμοί: $(x \geq 0 \text{ ΚΑΙ } y \geq 0) \text{ Η } (x < 0 \text{ ΚΑΙ } y < 0)$
ή $x * y > 0$ αν x και $y \neq 0$

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M14. Μετατροπή: Αν πολλαπλής επιλογής \leftrightarrow σύνολο από Αν-ΤέλοςΑν

Αν $(x < 0)$ τότε
Γράψε 1
Αλλιώς Αν $(x < 1000)$ τότε
Γράψε 2
Αλλιώς Αν $(x < 2000$ ΚΑΙ $x <> 1500)$ τότε
Γράψε 3
Αλλιώς
Γράψε 4
ΤέλοςΑν

Αν $(x < 0)$ τότε
Γράψε 1
ΤέλοςΑν
Αν $(x \geq 0$ ΚΑΙ $x < 1000)$ τότε
Γράψε 2
ΤέλοςΑν
Αν $(x \geq 1000$ ΚΑΙ $x < 2000$ ΚΑΙ $x <> 1500)$ τότε
Γράψε 3
ΤέλοςΑν
Αν $(x \geq 2000$ Η $x = 1500)$ τότε
Γράψε 4
ΤέλοςΑν

Διάβασε x
Αν $(x < 0)$ τότε
 $x \leftarrow x + 10$
Αλλιώς Αν $(x < 100)$ τότε
 $x \leftarrow x + 5$
Αλλιώς
 $x \leftarrow x + 1$
ΤέλοςΑν

Διάβασε x
Αν $(x < 0)$ τότε
 $x \leftarrow x + 10$
ΤέλοςΑν
Αν $(x \geq 0$ ΚΑΙ $x < 100)$ τότε
 $x \leftarrow x + 5$
ΤέλοςΑν
Αν $(x \geq 100)$ τότε
 $x \leftarrow x + 1$
ΤέλοςΑν
! Είναι σωστό;

Διάβασε x
 $ax \leftarrow x$
Αν $(ax < 0)$ τότε
 $x \leftarrow x + 10$
ΤέλοςΑν
Αν $(ax \geq 0$ ΚΑΙ $ax < 100)$ τότε
 $x \leftarrow x + 5$
ΤέλοςΑν
Αν $(ax \geq 100)$ τότε
 $x \leftarrow x + 1$
ΤέλοςΑν

Διάβασε x
 $\Sigma 1 \leftarrow x < 0$
 $\Sigma 2 \leftarrow x \geq 0$ ΚΑΙ $x < 100$
 $\Sigma 3 \leftarrow x \geq 100$
Αν $(\Sigma 1)$ τότε
 $x \leftarrow x + 10$
ΤέλοςΑν
Αν $(\Sigma 2)$ τότε
 $x \leftarrow x + 5$
ΤέλοςΑν
Αν $(\Sigma 3)$ τότε
 $x \leftarrow x + 1$
ΤέλοςΑν

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M14. Μετατροπή: Αν πολλαπλής επιλογής \leftrightarrow σύνολο από Αν-ΤέλοςΑν

```
ΔΙΑΒΑΣΕ x
ΑΝ x < 0 ΤΟΤΕ
  ΓΡΑΨΕ '@'
ΤΕΛΟΣ_ΑΝ
ΑΝ x < 10 ΤΟΤΕ
  ΓΡΑΨΕ '#'
ΤΕΛΟΣ_ΑΝ
ΑΝ x < 20 ΤΟΤΕ
  ΓΡΑΨΕ '$'
ΤΕΛΟΣ_ΑΝ
```




```
ΔΙΑΒΑΣΕ x
ΑΝ x < 0 ΤΟΤΕ
  ΓΡΑΨΕ '@'
  ΓΡΑΨΕ '#'
  ΓΡΑΨΕ '$'
ΑΛΛΙΩΣ_ΑΝ x < 10 ΤΟΤΕ
  ΓΡΑΨΕ '#'
  ΓΡΑΨΕ '$'
ΑΛΛΙΩΣ_ΑΝ x < 20 ΤΟΤΕ
  ΓΡΑΨΕ '$'
ΤΕΛΟΣ_ΑΝ
```

```
Αν x > 0 τότε
  x <-- x - 3
ΤέλοςΑν
Αν x <= 5 τότε
  x <-- 2*x
ΤέλοςΑν
```



```
Αν x <= 0 τότε
  x <-- 2*x
ΑλλιώςΑν x <= 8 τότε
  x <-- x - 3
  x <-- 2*x
Αλλιώς
  x <-- x - 3
ΤέλοςΑν
```



Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M15. Κατάργηση των ΚΑΙ, Η, ΟΧΙ

Αν (Σ1 ΚΑΙ Σ2) τότε
εντολές
ΤέλοςΑν

Σ1	Σ2	Εκτελούνται
A	A	εντολές
A	Ψ	-
Ψ	A	-
Ψ	Ψ	-

Αν (Σ1) τότε
Αν (Σ2) τότε
εντολές
ΤέλοςΑν
ΤέλοςΑν

Αν (Σ1 Η Σ2) τότε
εντολές
ΤέλοςΑν

Σ1	Σ2	Εκτελούνται
A	A	εντολές
A	Ψ	εντολές
Ψ	A	εντολές
Ψ	Ψ	-

Αν (Σ1) τότε
εντολές
ΑλλιώςΑν (Σ2) τότε
εντολές
ΤέλοςΑν

ή
Αν (Σ1) τότε
εντολές
ΤέλοςΑν
Αν (Σ2) τότε
εντολές
ΤέλοςΑν
! Είναι σωστό;

ή
Αν (Σ1) τότε
εντολές
ΤέλοςΑν
Αν (Σ2 ΚΑΙ ΟΧΙ Σ1) τότε
εντολές
ΤέλοςΑν

Γενικά:
Αν (Σ1 ΚΑΙ Σ2 ΚΑΙ ... ΚΑΙ Σν) τότε
εντολές
ΤέλοςΑν
είναι ισοδύναμο με:
Αν (Σ1) τότε
Αν (Σ2) τότε
...
Αν (Σν) τότε
εντολές
ΤέλοςΑν

...
ΤέλοςΑν
ΤέλοςΑν

Γενικά:
Αν (Σ1 Η Σ2 Η ... Η Σν) τότε
εντολές
ΤέλοςΑν
είναι ισοδύναμο με:
Αν (Σ1) τότε
εντολές
ΑλλιώςΑν (Σ2) τότε
εντολές

...
ΑλλιώςΑν (Σν) τότε
εντολές
ΤέλοςΑν

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M15. Κατάργηση των ΚΑΙ, Η, ΟΧΙ

Αν (Σ1 ΚΑΙ Σ2) τότε
εντολές1
Αλλιώς
εντολές2
ΤέλοςΑν

Σ1	Σ2	Εκτελούνται
A	A	εντολές1
A	Ψ	εντολές2
Ψ	A	εντολές2
Ψ	Ψ	εντολές2

Αν (Σ1) τότε
Αν (Σ2) τότε
εντολές1
Αλλιώς
εντολές2
ΤέλοςΑν
Αλλιώς
εντολές2
ΤέλοςΑν

Αν (Σ1 Η Σ2) τότε
εντολές1
Αλλιώς
εντολές2
ΤέλοςΑν

Σ1	Σ2	Εκτελούνται
A	A	εντολές1
A	Ψ	εντολές1
Ψ	A	εντολές1
Ψ	Ψ	εντολές2

Αν (Σ1) τότε
εντολές1
ΑλλιώςΑν (Σ2) τότε
εντολές1
Αλλιώς
εντολές2
ΤέλοςΑν

ή
Αν (Σ1) τότε
εντολές1
ΤέλοςΑν
Αν (ΟΧΙ Σ1 ΚΑΙ Σ2) τότε
εντολές1
ΤέλοςΑν
Αν (ΟΧΙ Σ1 ΚΑΙ ΟΧΙ Σ2) τότε
εντολές2
ΤέλοςΑν

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M15. Κατάργηση των ΚΑΙ, Η, ΟΧΙ

Αν (Σ1 Η Σ2 ΚΑΙ Σ3) τότε
εντολής1
ΤέλοςΑν

Σ1	Σ2	Σ3	Εκτελούνται
A	A	A	εντολής1
A	A	Ψ	εντολής1
A	Ψ	A	εντολής1
A	Ψ	Ψ	εντολής1
Ψ	A	A	εντολής1
Ψ	A	Ψ	-
Ψ	Ψ	A	-
Ψ	Ψ	Ψ	-

Αν (Σ1) τότε
εντολής1
ΑλλιώςΑν (Σ2) τότε
Αν (Σ3) τότε
εντολής1
ΤέλοςΑν
ΤέλοςΑν

Αν (Σ1 Η Σ2 ΚΑΙ Σ3) τότε
εντολής1
Αλλιώς
εντολής2
ΤέλοςΑν

Σ1	Σ2	Σ3	Εκτελούνται
A	A	A	εντολής1
A	A	Ψ	εντολής1
A	Ψ	A	εντολής1
A	Ψ	Ψ	εντολής1
Ψ	A	A	εντολής1
Ψ	A	Ψ	εντολής2
Ψ	Ψ	A	εντολής2
Ψ	Ψ	Ψ	εντολής2

Αν (Σ1) τότε
εντολής1
ΑλλιώςΑν (Σ2) τότε
Αν (Σ3) τότε
εντολής1
Αλλιώς
εντολής2
ΤέλοςΑν
Αλλιώς
εντολής2
ΤέλοςΑν

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M15. Κατάργηση των ΚΑΙ, Η, ΟΧΙ

Αν (Σ1 ΚΑΙ (Σ2 Η Σ3)) τότε
εντολές1
ΤέλοςΑν

Σ1	Σ2	Σ3	Εκτελούνται
A	A	A	εντολές1
A	A	Ψ	εντολές1
A	Ψ	A	εντολές1
A	Ψ	Ψ	-
Ψ	A	A	-
Ψ	A	Ψ	-
Ψ	Ψ	A	-
Ψ	Ψ	Ψ	-

Αν (Σ1) τότε
Αν (Σ2) τότε
εντολές1
ΑλλιώςΑν (Σ3) τότε
εντολές1
ΤέλοςΑν
ΤέλοςΑν

Αν (Σ1 ΚΑΙ (Σ2 Η Σ3)) τότε
εντολές1
Αλλιώς
εντολές2
ΤέλοςΑν

Σ1	Σ2	Σ3	Εκτελούνται
A	A	A	εντολές1
A	A	Ψ	εντολές1
A	Ψ	A	εντολές1
A	Ψ	Ψ	εντολές2
Ψ	A	A	εντολές2
Ψ	A	Ψ	εντολές2
Ψ	Ψ	A	εντολές2
Ψ	Ψ	Ψ	εντολές2

Αν (Σ1) τότε
Αν (Σ2) τότε
εντολές1
ΑλλιώςΑν (Σ3) τότε
εντολές1
Αλλιώς
εντολές2
ΤέλοςΑν
Αλλιώς
εντολές2
ΤέλοςΑν

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

Διάφορα (απλούστευση λογικών εκφράσεων)

Δώστε ισοδύναμες λογικές τιμές (Αληθής/Ψευδής) ή απλούστερες λογικές εκφράσεις για τα παρακάτω:

$x \geq 9 \text{ ΚΑΙ } x < 7$	Ψευδής	$x > 9 \text{ Η } x = 9$	\Leftrightarrow	$x \geq 9$
$y <> 5 \text{ Η } y <> 8$	Αληθής			
$z <> 11 \text{ Η } z \geq -1$	Αληθής			
$\varphi = 1 \text{ Η } \varphi <> 1$	Αληθής	$\Sigma \text{ ΚΑΙ ΟΧΙ } \Sigma = \text{Ψευδής}$		
$\lambda = 1 \text{ Η } 2 = 2$	Αληθής	$\Sigma \text{ Η ΟΧΙ } \Sigma = \text{Αληθής}$		
$\mu > 0 \text{ ΚΑΙ } 2 > 3$	Ψευδής			
$\rho > 1 \text{ ΚΑΙ } \rho \geq 4$	$\rho \geq 4$			
$\tau < 0 \text{ ΚΑΙ } \tau < 7$	$\tau < 0$			
$\Sigma = \text{Αληθής} \Leftrightarrow \Sigma \Leftrightarrow \Sigma <> \text{Ψευδής}$		$\text{ΟΧΙ } \Sigma \Leftrightarrow \Sigma = \text{Ψευδής} \Leftrightarrow \Sigma <> \text{Αληθής}$		
$x \geq y \text{ ΚΑΙ } y \geq x \Leftrightarrow x = y$		$x < y \text{ Η } x > y \Leftrightarrow x <> y$		

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

Διάφορα (έκφραση → λογική συνθήκη, συμπλήρωση κενών)

Έστω u το ύψος ενός ατόμου. Ποιες είναι οι αντίστοιχες λογικές συνθήκες;

όσοι έχουν ύψος μεταξύ του 1.80 και του 1.90

$u > 1.80$ ΚΑΙ $u < 1.90$

όσοι έχουν ύψος από 1.70 έως και 1.85

$u \geq 1.70$ ΚΑΙ $u \leq 1.85$

όσοι έχουν ύψος κάτω του 1.80 και
όσοι έχουν ύψος άνω του 1.90

$u < 1.80$ Η $u > 1.90$

Συμπληρώστε τα κενά με Αληθής/Ψευδής. Εάν ταιριάζουν και τα δύο, συμπληρώστε με τη λέξη Οτιδήποτε

1. Ψευδής Η ... = Αληθής

5. Αληθής ΚΑΙ ... = Ψευδής

2. Αληθής Η ... = ...

6. Αληθής Η Ψευδής ΚΑΙ Αληθής = ...

3. Ψευδής ΚΑΙ ... = ...

7. ΟΧΙ Ψευδής ΚΑΙ Αληθής Η ... = ...

4. Αληθής ΚΑΙ ... = Αληθής

1. Α 2. Ο, Α 3. Ο, Ψ 4. Α 5. Ψ 6. Α 7. Ο, Α

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

Διάφορα (Αν πολλαπλής επιλογής → διαστήματα τιμών)

Διάβασε x

Αν $(x < -5 \text{ Η } x > 80 \text{ ΚΑΙ } x \leq 140)$ τότε

Γράψε 1

Αλλιώς Αν $(x > 40 \text{ ΚΑΙ } x \leq 200)$ τότε

Γράψε 2

Αλλιώς

Γράψε 3


Αν $(x > 200)$ τότε

Γράψε 4

Τέλος Αν

Τέλος Αν

Για ποιες τιμές του x γράφει τις τιμές 1, 2, 3 και 4;



$$1: x \in (-\infty, -5) \cup (80, 140]$$

$$2: x \in (40, 80] \cup (140, 200]$$

$$3: x \in [-5, 40] \cup (200, \infty)$$

$$4: x \in (200, \infty)$$

Διάβασε ω, λ

$\lambda \leftarrow \lambda + 30$

Αν $\lambda \geq 60$ τότε

$\lambda \leftarrow \lambda - 60$

$\omega \leftarrow \omega + 1$

Αν $\omega = 24$ τότε

$\omega \leftarrow 0$

Τέλος Αν

Τέλος Αν


Γράψε ω, λ

Διάφορα (αλλαγή ώρας)

Πρόγραμμα που διαβάζει την τρέχουσα ώρα (ώρα: 0-23, λεπτά: 0-59) και εμφανίζει την ώρα μετά από 30'. π.χ. 10:15 → 10:45, 10:50 → 11:20, 23:50 → 0:20

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

Διάφορα (απαλοιφή κώδικα "κοινού παράγοντα" σε δομή Αν)



Δομή επιλογής – Μεθοδολογίες Ασκήσεων

Διάφορα (ισοδύναμος κώδικας)

Αν $(x \bmod 2 = 0)$ τότε
Γράψε x
Αλλιώς
Γράψε $-x$
ΤέλοςΑν



Γράψε $(-1)^{x*x} \dots$

ή

Γράψε $(-1)^{(x \bmod 2)*x}$

Αν $(x \bmod 2 = 0)$ τότε
Γράψε 100
Αλλιώς
Γράψε 99
ΤέλοςΑν



Γράψε $100 - A_T(x \bmod 2)$

Αν $(x \bmod 2 = 0)$ τότε
Γράψε 1
Αλλιώς
Γράψε x
ΤέλοςΑν



Γράψε $x^{(x \bmod 2)}$

Οι ακέραιοι x, y είναι
και οι δύο άρτιοι ή και
οι δύο περιπτοί



$x \bmod 2 = 0$ ΚΑΙ $y \bmod 2 = 0$ Η $x \bmod 2 \neq 0$ ΚΑΙ $y \bmod 2 \neq 0$



$x \bmod 2 = y \bmod 2$



$(x + y) \bmod 2 = 0$

Αν Σ τότε
 $\lambda \leftarrow$ Αληθής
Αλλιώς
 $\lambda \leftarrow$ Ψευδής
ΤέλοςΑν



$\lambda \leftarrow \Sigma$

Αν Σ τότε
 $\lambda \leftarrow$ Ψευδής
Αλλιώς
 $\lambda \leftarrow$ Αληθής
ΤέλοςΑν



Αν $(x \bmod 2 = 0)$ τότε
 $y \leftarrow$ Αληθής
Αλλιώς
 $y \leftarrow$ Ψευδής
ΤέλοςΑν

$\lambda \leftarrow$ ΟΧΙ Σ



$y \leftarrow x \bmod 2 = 0$

Αν $(x \bmod 2 = 0)$ τότε
 $y \leftarrow 1$
Αλλιώς
 $y \leftarrow 2$
ΤέλοςΑν



$y \leftarrow 2$
Αν $(x \bmod 2 = 0)$ τότε
 $y \leftarrow 1$
ΤέλοςΑν

ή

$y \leftarrow 1$
Αν $(x \bmod 2 \neq 0)$ τότε
 $y \leftarrow 2$
ΤέλοςΑν

Αν $(x > 0)$ τότε
 $y \leftarrow x$
Αλλιώς
 $y \leftarrow -x$
ΤέλοςΑν



$y \leftarrow A_T(x)$


(§ 2.4.5 - § 8.2) Δομή Επανάληψης

1. Δομή Όσο - επανάλαβε

Σύνταξη

```


ΟΣΟ συνθήκη ΕΠΑΝΑΛΑΒΕ
  εντολή-1
  εντολή-2
  ...
  εντολή-ν
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
    
```



Λειτουργία

Ελέγχεται η συνθήκη και αν είναι Αληθής, εκτελούνται οι εντολές που βρίσκονται ανάμεσα στις **ΟΣΟ_ΕΠΑΝΑΛΑΒΕ** και **ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ**. Στη συνέχεια ελέγχεται πάλι η συνθήκη και αν ισχύει, εκτελούνται πάλι οι ίδιες εντολές. Όταν η λογική έκφραση γίνει Ψευδής, τότε σταματάει η επανάληψη και εκτελείται η εντολή μετά το **ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ**.

αριθμός επαναλήψεων $[0, \infty)$, αριθμός ελέγχου της συνθήκης $[1, \infty)$
 αριθμός ελέγχου της συνθήκης = αριθμός επαναλήψεων + 1
 περιπτώσεις χρήσης (άγνωστο πλήθος επαναλήψεων που μπορεί να είναι και 0).



π.χ. 1	π.χ. 2	π.χ. 3
α ← 5 Όσο (α ≤ 11) επανάλαβε α ← α + 2 Τέλος Επανάληψης	α ← -3 Όσο (α > -1) επανάλαβε α ← α + 4 Τέλος Επανάληψης	α ← 5 Όσο (α <> 10) επανάλαβε α ← α + 2 Τέλος Επανάληψης

π.χ. 1	π.χ. 2	π.χ. 3
4 επαναλήψεις	0 επαναλήψεις	∞ επαναλήψεις

(§ 2.4.5 - § 8.2) Δομή Επανάληψης


1. Δομή Όσο - επανάλαβε

Παράδειγμα:

Πρόγραμμα που διαβάζει ένα αρχικό κεφάλαιο κατάθεσης σε τράπεζα με ετήσιο επιτόκιο 4%. Εμφανίζει σε πόσα χρόνια το κεφάλαιο θα διπλασιασθεί.

Διάβασε κ
έτη \leftarrow 0
Όσο ($\kappa < 2 * \kappa$) επανάλαβε
 έτη \leftarrow έτη + 1
 $\kappa \leftarrow \kappa + 4/100 * \kappa$
ΤέλοςΕπανάληψης
Γράψε έτη
! Είναι σωστό;

Διάβασε κ
έτη \leftarrow 0
στόχος $\leftarrow 2 * \kappa$
Όσο ($\kappa < \text{στόχος}$) επανάλαβε
 έτη \leftarrow έτη + 1
 $\kappa \leftarrow \kappa + 4/100 * \kappa$
ΤέλοςΕπανάληψης
Γράψε έτη



(§ 2.4.5 - § 8.2) Δομή Επανάληψης

2. Δομή ΜέχριςΌτου

Σύνταξη

ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ


εντολή-1

εντολή-2

...

εντολή-n

ΜΕΧΡΙΣ_ΟΤΟΥ λογική-έκφραση



Λειτουργία

Εκτελούνται οι εντολές μεταξύ των **ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ** και **ΜΕΧΡΙΣ_ΟΤΟΥ**. Στη συνέχεια ελέγχεται η λογική έκφραση και αν δεν ισχύει (είναι ψευδής), τότε οι εντολές που βρίσκονται ανάμεσα στις **ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ** και **ΜΕΧΡΙΣ_ΟΤΟΥ**, εκτελούνται πάλι. Ελέγχεται ξανά η λογική έκφραση και αν δεν ισχύει, επαναλαμβάνεται η εκτέλεση των ίδιων εντολών.

Όταν η λογική έκφραση γίνει Αληθής τότε σταματάει η επανάληψη και εκτελείται η εντολή μετά από την **ΜΕΧΡΙΣ_ΟΤΟΥ**.

αριθμός επαναλήψεων $[1, \infty)$, αριθμός ελέγχου της συνθήκης $[1, \infty)$
αριθμός ελέγχου της συνθήκης = αριθμός επαναλήψεων
περιπτώσεις χρήσης (άγνωστο πλήθος επαναλήψεων που είναι τουλάχιστον 1).

π.χ. 1	π.χ. 2	π.χ. 3
$\alpha \leftarrow 5$ ΑρχήΕπανάληψης $\alpha \leftarrow \alpha + 2$ ΜέχριςΌτου ($\alpha > 11$)	$\alpha \leftarrow -3$ ΑρχήΕπανάληψης $\alpha \leftarrow \alpha + 4$ ΜέχριςΌτου ($\alpha \geq 1$)	$\alpha \leftarrow 5$ ΑρχήΕπανάληψης $\alpha \leftarrow \alpha + 2$ ΜέχριςΌτου ($\alpha = 12$)

π.χ. 1	π.χ. 2	π.χ. 3
4 επαναλήψεις	1 επανάληψη	∞ επαναλήψεις

(§ 2.4.5 - § 8.2) Δομή Επανάληψης

2. Δομή ΜέχριςΌτου

Πρόγραμμα που διαβάζει επαναληπτικά θετικές ακέραιες τιμές και τις χαρακτηρίζει ως άρτιο/περιττό. Σε κάθε επανάληψη να ρωτάει τον χρήστη εάν επιθυμεί να συνεχίσει (ΝΑΙ/ΟΧΙ)

ΠΡΟΓΡΑΜΜΑ test

ΜΕΤΑΒΛΗΤΕΣ

ΑΚΕΡΑΙΕΣ: x

ΧΑΡΑΚΤΗΡΕΣ: ap

ΑΡΧΗ

ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ

ΔΙΑΒΑΣΕ x

ΑΝ $x \bmod 2 = 0$ ΤΟΤΕ

ΓΡΑΨΕ 'Άρτιος'

ΑΛΛΙΩΣ

ΓΡΑΨΕ 'Περιττός'


ΤΕΛΟΣ_ΑΝ

ΓΡΑΨΕ 'Συνέχεια (ΝΑΙ/ΟΧΙ);'

ΔΙΑΒΑΣΕ ap

ΜΕΧΡΙΣ_ΟΤΟΥ (ap = 'ΟΧΙ')

ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ



(§ 2.4.5 - § 8.2) Δομή Επανάληψης

For Loop

3. Δομή Για-από-μέχρι

Σύνταξη:

για <μετρητής> από <αρχική> μέχρι <τελική> [μεβήμα β]

<εντολές>

τέλοςΕπανάληψης

- όταν το βήμα δεν αναγράφεται, εννοείται το 1

ΔΙΑΓΡΑΜΜΑ ΡΟΗΣ	ΚΩΔΙΚΑΣ	ΔΙΑΓΡΑΜΜΑ ΡΟΗΣ	ΚΩΔΙΚΑΣ
<pre>graph TD; A[μ <-- α] --> B{μ <= τ}; B -- NAI --> C[εντολές]; C --> D[μ <-- μ + β]; D --> B; B -- OXI --> E[];</pre>	Για μ από α μέχρι τ μεβήμα β ! β > 0 Εντολές ΤέλοςΕπανάληψης	<pre>graph TD; A[μ <-- α] --> B{μ >= τ}; B -- NAI --> C[εντολές]; C --> D[μ <-- μ + β]; D --> B; B -- OXI --> E[];</pre>	Για μ από α μέχρι τ μεβήμα β ! β < 0 Εντολές ΤέλοςΕπανάληψης



Εντός της Για **δεν** επιτρέπεται η τροποποίηση της τιμής του μετρητή

(§ 2.4.5 - § 8.2) Δομή Επανάληψης

3. Δομή Για-από-μέχρι

Λειτουργία:

Ο ΗΥ "σαρώνει" μέσω του μετρητή το διάστημα:



και σε κάθε επανάληψη, εκτελεί τις εντολές.

Πιθανός αριθμός επαναλήψεων: $[0, +\infty)$

βήμα = 0 $\Rightarrow \infty$ επαναλήψεις

$\beta \neq 0$ και αρχική = τελική $\Rightarrow 1$ επανάληψη

$\beta > 0$ και αρχική > τελική ή

$\beta < 0$ και αρχική < τελική $\Rightarrow 0$ επαναλήψεις

σε κάθε άλλη περίπτωση, επαναλήψεις =

$$A_M\left(\left\lceil \frac{\text{τελική} - \text{αρχική}}{\text{βήμα}} \right\rceil + 1\right)$$

Τελική τιμή του μετρητή = η τιμή με την οποία "έσπασε" για 1η φορά το φράγμα του μέχρι

Χρήση: όταν έχω επανάληψη **ΓΝΩΣΤΟΥ** πλήθους επαναλήψεων.

(§ 2.4.5 - § 8.2) Δομή Επανάληψης

3. Δομή Για-από-μέχρι, παραδείγματα

Εμφάνιση των 100 πρώτων ακεραίων:
για x από 1 μέχρι 100

Γράψε x

ΤέλοςΕπανάληψης

Εμφάνιση των αριθμών: 50, 49, ... , 32, 31, 30
για y από 50 μέχρι 30 μεβήμα -1

Γράψε y

ΤέλοςΕπανάληψης

Εμφάνιση των αριθμών: 1.7, 1.8, ... , 9.7, 9.8
για z από 1.7 μέχρι 9.8 μεβήμα 0.1

Γράψε z

ΤέλοςΕπανάληψης

Εμφάνιση της λέξης "Καλημέρα" χίλιες φορές
για κ από 1 μέχρι 1000

Γράψε "Καλημέρα"

ΤέλοςΕπανάληψης

Εμφάνιση των 100 πρώτων
πολλαπλασίων του 7 :

για κ από 7 μέχρι 700 μεβήμα 7

Γράψε κ

ΤέλοςΕπανάληψης

για κ από 1 μέχρι 100

Γράψε ... 7*κ

ΤέλοςΕπανάληψης



Υπολογισμός του αθροίσματος
 $S = 5 + 10 + 15 + \dots + 500$:

$S \leftarrow 0$

για x από 5 μέχρι 500 μεβήμα 5

$S \leftarrow S + x$

ΤέλοςΕπανάληψης

Γράψε S

Υπολογισμός του γινομένου $\Gamma = 3 \times 6 \times 9 \times 12 \times \dots \times 300$:

$\Gamma \leftarrow 1$

για x από 3 μέχρι 300 μεβήμα 3

$\Gamma \leftarrow \Gamma * x$

ΤέλοςΕπανάληψης

Γράψε Γ

Υπολογισμός του αθροίσματος
 $S = 1^2 + 3^2 + 5^2 + \dots + 99^2$:
 $S \leftarrow 0$
για x από 1 μέχρι 99 μεβήμα 2
 $S \leftarrow S + x^2$
ΤέλοςΕπανάληψης
Γράψε S

Υπολογισμός της παράστασης:

$$S = \frac{1^2 + 3^2 + 5^2 + \dots + 99^2}{2^3 + 4^3 + 6^3 + \dots + 100^3}$$

$S1 \leftarrow 0$
 $S2 \leftarrow 0$
για x από 1 μέχρι 99 μεβήμα 2
 $S1 \leftarrow S1 + x^2$
 $S2 \leftarrow S2 + (x+1)^3$
ΤέλοςΕπανάληψης
 $S \leftarrow S1 / S2$
Γράψε S

Υπολογισμός του a^β , $\beta \in \mathbb{Z}_+$:
Διάβασε a, β
 $\Delta \leftarrow 1$
για x από 1 μέχρι β
 $\Delta \leftarrow \Delta * a$
ΤέλοςΕπανάληψης
Γράψε Δ

Υπολογισμός της παράστασης:
 $S = 1^{10} + 2^9 + 3^8 + \dots + 9^2 + 10^1$
 $S \leftarrow 0$
 $\Delta \leftarrow 10$
για x από 1 μέχρι 10
 $S \leftarrow S + x^\Delta$
 $\Delta \leftarrow \Delta - 1$
ΤέλοςΕπανάληψης
Γράψε S


Υπολογισμός του a^β , $\beta \in \mathbb{Z}$:
Διάβασε a, β
 $\Delta \leftarrow 1$
για x από 1 μέχρι $A_T(\beta)$
 $\Delta \leftarrow \Delta * a$
ΤέλοςΕπανάληψης
Αν ($\beta < 0$) τότε
 $\Delta \leftarrow 1/\Delta$
ΤέλοςΑν
Γράψε Δ

Εμφάνιση των ακεραίων
του $[a,b]$ ίσια και
ανάποδα ταυτόχρονα:
a b, a+1 b-1, ..., b a

ΓΙΑ x ΑΠΟ a ΜΕΧΡΙ b
ΓΡΑΨΕ x, a+b-x
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

ποιό το πλήθος των επαναλήψεων και τί εμφανίζουν:

1. για κ από 5 μέχρι 35 μεβήμα 7, Γράψε κ
2. για λ από 14 μέχρι 31 μεβήμα 0 , Γράψε λ
3. για μ από 15 μέχρι 4 μεβήμα 2 , Γράψε μ
4. για ν από 11 μέχρι 23 μεβήμα -3 , Γράψε ν
5. για λ από 11 μέχρι 11 μεβήμα 4 , Γράψε λ
6. για x από 314 μέχρι 5729 μεβήμα 7 (τύπος \Rightarrow 774 επαναλήψεις)
7. $T \leftarrow 1$; για x από 1 μέχρι T; Γράψε x; $T \leftarrow T + 1$; ΤέλοςΕπανάληψης $\Rightarrow \infty$
8. για κ από -100 μέχρι 100 μεβήμα 0.1, Γράψε κ



ποιό το πλήθος των επαναλήψεων και τί εμφανίζει:

για x από 1 μέχρι 9 μεβήμα 2
για y από 11 μέχρι 2 μεβήμα -3
Γράψε x, y
ΤέλοςΕπανάληψης
ΤέλοςΕπανάληψης

Εμφάνιση: 1,-2,3,-4,...,99,-100

Εμφάνιση όλων των συνδυασμών ρίψης 2 ζαριών:

για x από 1 μέχρι 6
για y από 1 μέχρι 6
Γράψε x, y
ΤέλοςΕπανάληψης
ΤέλοςΕπανάληψης

Εμφάνιση όλων των ενδείξεων ενός ψηφιακού ρολογιού από την ώρα 0:0:0 έως και 23:59:59:
για ω από 0 μέχρι 23
για λ από 0 μέχρι 59
για δ από 0 μέχρι 59
Γράψε ω, ":", λ, ":", δ
ΤέλοςΕπανάληψης
ΤέλοςΕπανάληψης
ΤέλοςΕπανάληψης

για i από 1 μέχρι 100
Αν $(i \bmod 2 = 0)$ τότε
Γράψε -i
Αλλιώς
Γράψε i
ΤέλοςΑν
ΤέλοςΕπανάληψης

! ή
 $n \leftarrow 1$
για i από 1 μέχρι 100
Γράψε $n*i$
 $n \leftarrow -n$
ΤέλοςΕπανάληψης

! ή
για i από 1 μέχρι 99 μεβήμα 2
Γράψε i, $-(i+1)$
ΤέλοςΕπανάληψης

! ή
για i από 1 μέχρι 100
Γράψε $-(-1)^{(i \bmod 2)*i}$
ΤέλοςΕπανάληψης

Πότε τερματίζεται το παρακάτω:
για x από 1 μέχρι 100
Διάβασε x
ΤέλοςΕπανάληψης

Τι εμφανίζει το παρακάτω:
για x από 1 μέχρι 10
για x από 1 μέχρι 10
ΤέλοςΕπανάληψης
Γράψε x
ΤέλοςΕπανάληψης
Γράψε x

Εμφάνιση των 100 πρώτων
όρων της αναδρομικής
ακολουθίας:

$$a_n = \begin{cases} 5 & , n = 1 \\ 3 \cdot a_{n-1} + n - 2, & n \geq 2 \end{cases}$$

an ← 5
Γράψε an
για n από 2 μέχρι 100
an ← 3*an+n-2
Γράψε an
ΤέλοςΕπανάληψης

Όταν και εφόσον δοθεί τιμή > 99
(γενικά όμως: εντός της για δεν επιτρέπεται η τροποποίηση του μετρητή)

11, 12 (αν και παραβιάζεται ένας εκ των κανόνων των εμφωλευμένων βρόχων)


Εμφάνιση των τιμών της συνάρτησης:

$$f(x) = \frac{x + 9}{3 \cdot x^2 - 11 \cdot x + 5} \quad \forall x \in [-50, 50], x \in Z$$

για x από -50 μέχρι 50
par ← 3*x^2-11*x+5
Αν (par <> 0) τότε
Γράψε (x+9)/par
Αλλιώς
Γράψε 'αδύνατη'
ΤέλοςΑν
ΤέλοςΕπανάληψης

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M1. Μετατροπές δεδομένου επαναληπτικού κώδικα: Όσο ↔ Μέχρι ↔ Για



(1) Για → Όσο		
για x από 1 μέχρι 100 Γράψε x ΤέλοςΕπανάληψης	⇒	x ← 1 Όσο (x ≤ 100) επανάλαβε Γράψε x x ← x + 1 ΤέλοςΕπανάληψης

(2) Για → Μέχρι
Για → Όσο → Μέχρι

(3) Όσο → Μέχρι		
Όσο (Σ) επανάλαβε εντολές ΤέλοςΕπανάληψης	⇒	Αν (Σ) τότε ΑρχήΕπανάληψης εντολές ΜέχριςΌτου (ΟΧΙ Σ) ΤέλοςΑν

(5, 6) Όσο / Μέχρι → Για		
Η μετατροπή είναι δυνατή, εάν ισχύουν τα παρακάτω:		
<ul style="list-style-type: none"> • μία μεταβλητή ελέγχου (μετρητής) • γνωστή αρχική τιμή (από) • γνωστή τελική τιμή (μέχρι) • γνωστή και σταθερή μετατροπή του μετρητή (βήμα) 		

(4) Μέχρι → Όσο		
ΑρχήΕπανάληψης εντολές ΜέχριςΌτου (Σ)	⇒	εντολές Όσο (ΟΧΙ Σ) επανάλαβε εντολές ΤέλοςΕπανάληψης

(6) Μέχρι → Για
Μέχρι → Όσο → Για

Όσο → Για		
Διάβασε x Όσο (x <> 0) επανάλαβε εντολές Διάβασε x ΤέλοςΕπανάληψης	⇒	X

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M1. Μετατροπές δεδομένου επαναληπτικού κώδικα: Όσο \rightarrow Για

$\mu \leftarrow$ αρχική

Όσο ($\mu \leq$ τελική) επανάλαβε ! ή \geq

εντολές1

$\mu \leftarrow \mu + \beta$! ή $\mu \leftarrow \mu - \beta$

εντολές2

ΤέλοςΕπανάληψης

Μετατροπή σε για



για μ από αρχική μέχρι τελική μεβήμα β

εντολές1 ! όπου $\mu: \mu$

εντολές2 ! όπου $\mu: \mu + \beta$ ή $\mu - \beta$

ΤέλοςΕπανάληψης

π.χ.

$\mu \leftarrow 5$

Όσο ($\mu \leq 100$) επανάλαβε

Γράψε μ

$\mu \leftarrow \mu + 7$

Γράψε μ

ΤέλοςΕπανάληψης

Μετατροπή σε για



για μ από 5 μέχρι 100 μεβήμα 7

Γράψε μ

Γράψε $\mu + 7$

ΤέλοςΕπανάληψης

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M1. Μετατροπές δεδομένου επαναληπτικού κώδικα: Όσο \rightarrow Για

$x \leftarrow 1$

Όσο ($x < 10$) επανάλαβε

εντολές

$x \leftarrow x + 1$

ΤέλοςΕπανάληψης

Μετατροπή σε για



για x από 1 μέχρι 9

εντολές

ΤέλοςΕπανάληψης

$x \leftarrow 1.8$

Όσο ($x < 10$) επανάλαβε

εντολές

$x \leftarrow x + 0.1$

ΤέλοςΕπανάληψης

Μετατροπή σε για



για x από 1.8 μέχρι 9.9 μεβήμα 0.1

εντολές

ΤέλοςΕπανάληψης

$x \leftarrow 9.3$

Όσο ($x > 0$) επανάλαβε

εντολές

$x \leftarrow x - 0.1$

ΤέλοςΕπανάληψης

Μετατροπή σε για



για x από 9.3 μέχρι 0.1 μεβήμα -0.1

εντολές

ΤέλοςΕπανάληψης

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M2. Στατιστικά στοιχεία: ΜΟ, Sum, min, max, πλήθη, % ποσοστά

α) γνωστού πλήθους στοιχείων

π.χ. Πρόγραμμα που για τους 500 υπαλλήλους μιας εταιρείας, διαβάζει το μισθό τους και τον τομέα τους (Α/Β/Γ). Εμφανίζει: τον μέσο μισθό της εταιρείας, τους μέσους μισθούς ανά τομέα, τον μικρότερο και τον μεγαλύτερο μισθό, το % ποσοστό των υψηλόμισθων (> 2000 €)

$S \leftarrow 0$ $SA \leftarrow 0$ $SB \leftarrow 0$ $SG \leftarrow 0$

$PA \leftarrow 0$ $PB \leftarrow 0$ $PG \leftarrow 0$ $PY \leftarrow 0$

για i από 1 μέχρι 500

Διάβασε μ , τ

$S \leftarrow S + \mu$

Αν ($\tau = \text{"A"}$) τότε

$SA \leftarrow SA + \mu$

$PA \leftarrow PA + 1$

Αλλιώς Αν ($\tau = \text{"B"}$) τότε

$SB \leftarrow SB + \mu$

$PB \leftarrow PB + 1$

Αλλιώς

$SG \leftarrow SG + \mu$

$PG \leftarrow PG + 1$

ΤέλοςΑν

Αν ($\mu > 2000$) τότε

$PY \leftarrow PY + 1$

ΤέλοςΑν

Αν ($i = 1$) τότε

$min \leftarrow \mu$

$max \leftarrow \mu$

Αλλιώς

Αν ($\mu < min$) τότε

$min \leftarrow \mu$

ΤέλοςΑν

Αν ($\mu > max$) τότε

$max \leftarrow \mu$

ΤέλοςΑν

ΤέλοςΑν

ΤέλοςΕπανάληψης

$mo \leftarrow S / 500$

Γράψε mo , min , max , $PY/500*100$, "%"

Αν ($PA <> 0$) τότε

Γράψε SA / PA

ΤέλοςΑν

! το ίδιο για PB , PG

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M2. Στατιστικά στοιχεία: MO, Sum, min, max, πλήθη, % ποσοστά

β) αγνώστου πλήθους στοιχείων

π.χ1. Πρόγραμμα που διαβάζει επαναληπτικά τους βαθμούς μαθητών σε 1 διαγώνισμα (0-20) μέχρι να δοθεί ως βαθμός η τιμή -1. Εμφανίζει: στατιστικά...

Διάβασε β
Όσο ($\beta \neq -1$) επανάλαβε
! Στατιστικά...
Διάβασε β
ΤέλοςΕπανάληψης

Τιμή φρουρός: τιμή που ορίζεται από τον προγραμματιστή για τον τερματισμό μίας επαναληπτικής διαδικασίας. Είναι τέτοια, ώστε να μην είναι λογικά σωστή για το πρόβλημα.

ΑρχήΕπανάληψης
Διάβασε β
! Στατιστικά...
ΜέχριςΌτου ($\beta = -1$)
! Είναι σωστό;

ΑρχήΕπανάληψης
Διάβασε β
Αν ($\beta \neq -1$) τότε
! Στατιστικά...
ΤέλοςΑν
ΜέχριςΌτου ($\beta = -1$)

M2. Στατιστικά στοιχεία: ΜΟ, Sum, min, max, πλήθη, % ποσοστά

β) αγνώστου πλήθους στοιχείων

π.χ1. Πρόγραμμα που διαβάζει επαναληπτικά τους βαθμούς μαθητών σε 1 διαγώνισμα (0-20) μέχρι να δοθεί ως βαθμός η τιμή -1. Εμφανίζει: τον μέσο όρο βαθμολογίας των μαθητών, τον μικρότερο και τον μεγαλύτερο βαθμό, το % ποσοστό των άριστων (> 18)

```
Σ ← 0 Π ← 0 ΠΑ ← 0
```

```
min ← 21 max ← -1
```

```
Διάβασε β
```

```
Όσο (β <> -1) επανάλαβε
```

```
Σ ← Σ + β
```

```
Π ← Π + 1
```

```
Αν (β > max) τότε
```

```
max ← β
```

```
ΤέλοςΑν
```

```
Αν (β < min) τότε
```

```
min ← β
```

```
ΤέλοςΑν
```

```
Αν (β > 18) τότε
```

```
ΠΑ ← ΠΑ + 1
```

```
ΤέλοςΑν
```

```
Διάβασε β
```

```
ΤέλοςΕπανάληψης
```

```
Αν (Π <> 0) τότε
```

```
ΜΟ ← Σ / Π
```

```
Γράψε ΜΟ, min, max, ΠΑ/Π*100, '%'
```

```
Αλλιώς
```

```
Γράψε 'Δεν δόθηκαν βαθμοί'
```

```
ΤέλοςΑν
```

β) αγνώστου πλήθους στοιχείων

π.χ2. Πρόγραμμα που διαβάζει το όνομα, την ηλικία και τον μισθό υπαλλήλων μέχρι να δοθεί ως όνομα λέξη 'ΤΕΛΟΣ'. Εμφανίζει το όνομα του μεγαλύτερου σε ηλικία (μόνο ένας) και τον μέσο μισθό.

```
max <-- 0
```

```
s <-- 0
```

```
p <-- 0
```

```
Διάβασε όνομα
```

```
Όσο όνομα <> 'ΤΕΛΟΣ' επανάλαβε
```

```
Διάβασε ηλικία, μισθός
```

```
p <-- p + 1
```

```
s <-- s + μισθός
```

```
Αν ηλικία > max τότε
```

```
max <-- ηλικία
```

```
maxΟν <-- όνομα
```

```
ΤέλοςΑν
```

```
Διάβασε όνομα
```

```
ΤέλοςΕπανάληψης
```

```
Αν p <> 0 τότε
```

```
Γράψε maxΟν, s/p
```

```
Αλλιώς
```

```
Γράψε 'κανένας'
```

```
ΤέλοςΑν
```

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M2. Στατιστικά στοιχεία: MO, Sum, min, max, πλήθη, % ποσοστά

β) αγνώστου πλήθους στοιχείων

π.χ3. Πρόγραμμα που διαβάζει επαναληπτικά τους βαθμούς μαθητών σε 1 διαγώνισμα (0-20) μέχρι να δοθεί ως βαθμός η τιμή -1. Εμφανίζει τους 2 μεγαλύτερους βαθμούς. Θεωρείστε ότι θα δοθούν 2 διαφορετικοί βαθμοί

max1 ← -1 max2 ← -1

Διάβασε β

Όσο (β <> -1) επανάλαβε

Αν (β > max1) τότε

max2 ← max1

max1 ← β

Αλλιώς Αν (β > max2) τότε

max2 ← β

Τέλος Αν

Διάβασε β

Τέλος Επανάληψης

Γράψε max1, max2

π.χ4. Πρόγραμμα που διαβάζει επαναληπτικά τους βαθμούς μαθητών σε 1 διαγώνισμα (0-20) μέχρι να δοθεί ως βαθμός η τιμή -1. Εμφανίζει το μεγαλύτερο βαθμό και το πλήθος των βαθμών που είναι ίσοι με αυτόν. Θεωρείστε ότι θα δοθεί ένας βαθμός

max ← -1

Πmax ← 0

Διάβασε β

Όσο (β <> -1) επανάλαβε

Αν (β > max) τότε

max ← β

Πmax ← 1

Αλλιώς Αν (β = max) τότε

Πmax ← Πmax + 1

Τέλος Αν

Διάβασε β

Τέλος Επανάληψης

Γράψε max, Πmax

π.χ5. Πρόγραμμα που διαβάζει επαναληπτικά τιμές μέχρι να δοθεί ως βαθμός η τιμή 0. Εμφανίζει το πλήθος των «αιχμών». Αιχμή θεωρείται κάθε τιμή που είναι μεγαλύτερη από την προηγούμενη και την επόμενη της. Να θεωρήσετε ότι θα δοθούν 3 τιμές διαφορετικές του 0.

ΠΡΟΓΡΑΜΜΑ aixmes
ΜΕΤΑΒΛΗΤΕΣ



ΑΚΕΡΑΙΕΣ: a,b,c,pa

ΑΡΧΗ

ΓΡΑΨΕ 'Give 1st, 2nd and 3rd number'

ΔΙΑΒΑΣΕ a,b,c

pa <-- 0

ΟΣΟ c <> 0 ΕΠΑΝΑΛΑΒΕ

ΑΝ b > a ΚΑΙ b > c ΤΟΤΕ

pa <-- pa + 1

ΤΕΛΟΣ_ΑΝ

a <-- b

b <-- c

ΓΡΑΨΕ 'Give next number'

ΔΙΑΒΑΣΕ c

ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

ΓΡΑΨΕ pa

ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ

π.χ6. Πρόγραμμα που διαβάζει επαναληπτικά αριθμητικές τιμές μέχρι να δοθεί τιμή που ισούται με το άθροισμα της προηγούμενης και της προπροηγούμενης τιμής. Να εμφανίζει το πλήθος των τιμών που δόθηκαν.

ΔΙΑΒΑΣΕ a, b, c

p <-- 3

ΟΣΟ c <> a + b ΕΠΑΝΑΛΑΒΕ

a <-- b

b <-- c

ΔΙΑΒΑΣΕ c

p <-- p + 1

ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

ΓΡΑΨΕ p

π.χ7. Πρόγραμμα που διαβάζει επαναληπτικά τον όροφο κλήσης (ακέραιος) ενός ασανσέρ μέχρι να δοθεί η τιμή 111. Να εμφανίζει το πλήθος των ορόφων που διένυσε συνολικά. Να θεωρηθεί ότι αρχικά το ασανσέρ βρίσκεται στο ισόγειο (0). π.χ. είσοδος: 3,1,5,-1,0,111 έξοδος: 3+2+4+6+1=16 όροφοι

S <--0

op <-- 0

Διάβασε x

'Όσο x <> 111 επανάλαβε

S <-- S + A_T(op-x)

op <-- x

Διάβασε x

ΤέλοςΕπανάληψης

Γράψε S

β) αγνώστου πλήθους στοιχείων

π.χ.8 Εισαγωγή ονομάτων και μισθών για τους υπαλλήλους μιας εταιρείας. Να διακόπτεται όταν δοθεί σαν όνομα η τιμή 'τέλος' ή δοθούν 10 υψηλόμισθοι (>2000€). Εμφάνιση του πιο καλά αμειβόμενου υπαλλήλου καθώς και τον μισθό του (μόνο ένας).

max ← -1

ΠΥ ← 0

Διάβασε on

Όσο (on <> 'τέλος' ΚΑΙ ΠΥ < 10) επανάλαβε

Διάβασε μ

Αν (μ > 2000) τότε

ΠΥ ← ΠΥ + 1

ΤέλοςΑν

Αν (μ > max) τότε

max ← μ

onmax ← on

ΤέλοςΑν

Αν (ΠΥ < 10) τότε

Διάβασε on

ΤέλοςΑν

ΤέλοςΕπανάληψης

Αν (max <> -1) τότε

Γράψε onmax, max

ΤέλοςΑν

M3. Τύποι μετατροπής για ένα διάστημα τιμών

π.χ. Πρόγραμμα που για κάθε θερμοκρασία κελσίου [-100, 100] μεβήμα 0.1, εμφανίζει την αντίστοιχη θερμοκρασία Fahrenheit $((F-32)/9 = C/5)$

για C από -100 μέχρι 100 μεβήμα 0.1

F ← $9 * C / 5 + 32$

Γράψε F

ΤέλοςΕπανάληψης

M4. Έλεγχος εγκυρότητας δεδομένων

π.χ. Εισαγωγή βαθμού στην 20-θμια κλίμακα με έλεγχο εγκυρότητας

1ος τρόπος

ΑρχήΕπανάληψης

Διάβασε β

ΜέχριςΌτου (β >= 0 ΚΑΙ β <=20)

2ος τρόπος

Διάβασε β

Όσο (β < 0 Η β > 20) επανάλαβε

Γράψε "Λάθος τιμή"

Διάβασε β

ΤέλοςΕπανάληψης

Για εισαγωγή πολλαπλών δεδομένων, μία ξεχωριστή ΜέχριςΌτου/Όσο για το καθένα

M5. Ποσότητα που μεταβάλλεται

α) για γνωστό πλήθος επαναλήψεων
π.χ. Πρόγραμμα που διαβάσει ένα αρχικό κεφάλαιο κατάθεσης σε τράπεζα και το % ετήσιο επιτόκιο της. Εμφανίζει το τελικό κεφάλαιο μετά από 30 χρόνια.

```
Διάβασε κ, επ
για i από 1 μέχρι 30
  κ ← κ + επ/100*κ
ΤέλοςΕπανάληψης
Γράψε κ
```

β) για άγνωστο πλήθος επαναλήψεων (2 ποσότητες)
π.χ.2 Δύο δέντρα έχουν σήμερα ύψος 2m και 3m αντίστοιχα. Ετησίως ψηλώνουν και 5% και 4% αντίστοιχα. Πρόγραμμα που υπολογίζει σε πόσα χρόνια το 1ο δέντρο θα ξεπεράσει σε ύψος το 2ο και τι ύψος θα έχουν τότε.

β) για άγνωστο πλήθος επαναλήψεων
π.χ.1 Έστω ότι σήμερα ζουν 40.000 θαλάσσιες χελώνες και ετησίως μειώνονται κατά 5,7%. Πρόγραμμα που υπολογίζει σε πόσα χρόνια θα γίνουν λιγότερες από 10.000.

```
X ← 40000
έτη ← 0
Όσο (X >= 10000) επανάλαβε
  X ← X - 5.7/100*X
  έτη ← έτη + 1
ΤέλοςΕπανάληψης
Γράψε έτη
```

```
h1 ← 2
h2 ← 3
eth ← 0
ΌΣΟ h1 <= h2 ΕΠΑΝΑΛΑΒΕ
  h1 ← h1 + 5/100*h1
  h2 ← h2 + 4/100*h2
  eth ← eth + 1
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

```
ΓΡΑΨΕ 'Σε ',eth,'χρόνια το 1ο δέντρο:',h1,'το 2ο δέντρο:',h2
```

! για εξαφάνιση του είδους

```
X ← 40000
έτη ← 0
Όσο (X > 0) επανάλαβε
  X ← X - 5.7/100*X
  έτη ← έτη + 1
ΤέλοςΕπανάληψης
Γράψε έτη
! Είναι σωστό;
```

Σε 43 χρόνια το 1ο δέντρο: 16.30 το 2ο δέντρο: 16.20

β) για άγνωστο πλήθος επαναλήψεων
 π.χ.3 Έστω ότι σήμερα είναι Δευτέρα και μία φωλιά
 περιέχει 500 κόκκους σιταριού. Κάθε μέρα τα
 μυρμήγκια βάζουν στη φωλιά 50 κόκκους. Κάθε
 Κυριακή τρώνε το 5% των κόκκων. Σε πόσες
 ημέρες θα έχουν μαζευτεί 10000 ή περισσότεροι
 κόκκοι;

```

k ← 500
days ← 0
Όσο (k < 10000) επανάλαβε
  days ← days + 1
  k ← k + 50
  Αν (days mod 7 = 0) τότε
    k ← k - 5/100*k
  ΤέλοςΑν
ΤέλοςΕπανάληψης
Γράψε days
  
```

M6. Ερώτηση προς τον χρήστη για συνέχεια (ΝΑΙ/ΟΧΙ)

π.χ. Πρόγραμμα που διαβάζει
 επαναληπτικά βαθμούς
 απολυτηρίων και τους
 χαρακτηρίζει ως: άριστος (>18)
 / πολύ καλός (>16) / άλλος.
 Στο τέλος κάθε επανάληψης να
 ρωτάει τον χρήστη εάν επιθυμεί
 να συνεχίσει (ΝΑΙ/ΟΧΙ).

```


ΑρχήΕπανάληψης
Διάβασε β
Αν (β > 18) τότε
  Γράψε 'άριστος'
ΑλλιώςΑν (β > 16) τότε
  Γράψε 'πολύ καλός'
Αλλιώς
  Γράψε 'άλλος'
ΤέλοςΑν
Γράψε 'Συνέχεια; (ΝΑΙ/ΟΧΙ)'
Διάβασε απ
ΜέχριςΌτου(απ = "ΟΧΙ")
  
```

! για ερώτηση από την αρχή
 Γράψε 'Συνέχεια; (ΝΑΙ/ΟΧΙ)'
 Διάβασε απ
 Όσο απ = 'ΝΑΙ' επανάλαβε
 Διάβασε β
 Αν (β > 18) τότε
 Γράψε 'άριστος'
 ΑλλιώςΑν (β > 16) τότε
 Γράψε 'πολύ καλός'
 Αλλιώς
 Γράψε 'άλλος'
 ΤέλοςΑν
 Γράψε 'Συνέχεια; (ΝΑΙ/ΟΧΙ)'
 Διάβασε απ
 Τέλοςεπανάληψης




(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων


M7. Μετατροπή : Αλγόριθμος ↔ ΔΡ




Αν (Συνθήκη) τότε
εντολές
ΤέλοςΑν



Αν (Συνθήκη) τότε
εντολές 1
Αλλιώς
εντολές 2
ΤέλοςΑν




Αν (Συνθήκη 1) τότε
εντολές 1
ΑλλιώςΑν (Συνθήκη 2) τότε
εντολές 2
...
ΑλλιώςΑν (Συνθήκη v) τότε
εντολές v
Αλλιώς
εντολές v+1
ΤέλοςΑν




Αν (Συνθήκη 1) τότε
εντολές 1
ΑλλιώςΑν (Συνθήκη 2) τότε
εντολές 2
...
ΑλλιώςΑν (Συνθήκη v) τότε
εντολές v
ΤέλοςΑν

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων


M7. Μετατροπή : Αλγόριθμος \leftrightarrow ΔΡ




Όσο (Συνθήκη) επανάλαιβε
εντολές
Τέλος Επανάληψης



Αρχή Επανάληψης
Εντολές
Μέχρις Ότου (Συνθήκη)




Για μ από α μέχρι τ μεβήμα β ! β > 0
Εντολές
Τέλος Επανάληψης




Για μ από α μέχρι τ μεβήμα β ! β < 0
Εντολές
Τέλος Επανάληψης

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων


M7. Μετατροπή : Αλγόριθμος ↔ ΔΡ



Αν (OXI Συνθήκη) τότε
εντολές
ΤέλοςΑν




Όσο (OXI Συνθήκη) επανάλαβε
εντολές
ΤέλοςΕπανάληψης



ΑρχήΕπανάληψης
εντολές
ΜέχριςΌτου (OXI Συνθήκη)

ή


εντολές
Όσο (Συνθήκη) επανάλαβε
εντολές
ΤέλοςΕπανάληψης



εντολές 1
Όσο (Συνθήκη) επανάλαβε
εντολές 2
εντολές 1
ΤέλοςΕπανάληψης

ή
ΑρχήΕπανάληψης
εντολές 1
Αν Συνθήκη τότε
εντολές 2
ΤέλοςΑν
ΜέχριςΌτου OXI Συνθήκη


ή
ΑρχήΕπανάληψης
εντολές 1
Αν OXI Συνθήκη τότε
εντολές 2
ΤέλοςΑν
ΜέχριςΌτου Συνθήκη




εντολές 1
Όσο (OXI Συνθήκη) επανάλαβε
εντολές 2
εντολές 1
ΤέλοςΕπανάληψης

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M7. Μετατροπή : Αλγόριθμος ↔ ΔΡ




stop ← Ψευδής
 Όσο (Σ1 ΚΑΙ stop = Ψευδής) επανάλαβε
 εντολές1
 Αν (Σ2) τότε
 εντολές2
 Αλλιώς
 εντολές3
 stop ← Αληθής
 ΤέλοςΑν
 ΤέλοςΕπανάληψης




Αν Σ1 τότε
 εντ1
 ΑλλιώςΑν Σ2 τότε
 εντ2
 Αλλιώς
 εντ3
 ΤέλοςΑν

ή

Αν Σ1 τότε
 εντ1
 Αλλιώς
 Αν Σ2 τότε
 εντ2
 Αλλιώς
 εντ3
 ΤέλοςΑν
 ΤέλοςΑν




Πρόγραμμα ΔΡ
 Μεταβλητές
 Ακέραιες: Α, Β, S
 Αρχή
 Διάβασε Α, Β
 S ← 0
 Όσο (B > 0) επανάλαβε
 Αν (B mod 2 = 1) τότε
 S ← S + A
 ΤέλοςΑν
 Α ← Α * 2
 Β ← Β div 2
 ΤέλοςΕπανάληψης
 Γράψε S
 ΤέλοςΠρογράμματος



Αλγόριθμος Μετατροπή
Αρχή_επανάληψης
 Διάβασε πλήθος
Μέχρις_ότου πλήθος > 0
 άθροισμα ← 0
 i ← 1
Όσο i <= πλήθος επανάλαβε
 Διάβασε αριθμός
 άθροισμα ← άθροισμα + αριθμός
 i ← i + 1
Τέλος_επανάληψης
 Εμφάνισε άθροισμα
Τέλος Μετατροπή

Αλγόριθμος Άσκηση
 Διάβασε x, n
 m ← n
 ζ ← x
 αποτέλεσμα ← 1
Όσο m > 0 επανάλαβε
 Όσο m mod 2 = 0 επανάλαβε
 m ← m div 2
 z ← z * ζ
Τέλος_επανάληψης
 m ← m - 1
 αποτέλεσμα ← αποτέλεσμα * ζ
Τέλος_επανάληψης
 Εκτύπωσε αποτέλεσμα
Τέλος Άσκηση



Αλγόριθμος Άσκηση

$\beta \leftarrow 3$

$\tau \leftarrow 0$

Για i από 1 μέχρι 4

$\beta \leftarrow \beta + 2$

Για j από β μέχρι i με_βήμα -2

$\tau \leftarrow \tau + j$

Τέλος επανάληψης

Τέλος επανάληψης

Εμφάνισε τ

Τέλος Άσκηση

M8. Επαναληπτικό μενού επιλογών με έξοδο

π.χ. Πρόγραμμα που με επαναληπτικό μενού επιλογών εκτελεί τις παρακάτω λειτουργίες: 1. Εμβαδό κύκλου 2. Εμβαδό τραπεζίου 3. Εμβαδό κυλίνδρου ($2\pi R^2 + 2\pi Rh$) 4. Έξοδος

Αρχή Επανάληψης

Γράψε "1. Εμβαδό κύκλου"

Γράψε "2. Εμβαδό τραπεζίου"

Γράψε "3. Εμβαδό κυλίνδρου"

Γράψε "4. Έξοδος"

Γράψε "Διάλεξε (1-4)"

Διάβασε επ

Αν (επ = 1) τότε

Διάβασε ρ

$E \leftarrow 3.14 * \rho^2$

Γράψε E

Αλλιώς Αν (επ = 2) τότε

Διάβασε β, B, u

$E \leftarrow (\beta + B) * u / 2$

Γράψε E

Αλλιώς Αν (επ = 3) τότε

Διάβασε R, h

$E \leftarrow 2 * 3.14 * R^2 + 2 * 3.14 * R * h$


Γράψε E

Αλλιώς Αν (επ <> 4) τότε

Γράψε "Λάθος επιλογή"

Τέλος Αν

Μέχρις Ότου (επ = 4)



M9. Πίνακας τιμών μεταβλητών

Γράψε '!' για x από 1 μέχρι 11
Γράψε '@' για y από 1 μέχρι 20
Γράψε '#'
Αν $y \bmod 2 = 0$ τότε
Γράψε '^'
ΤέλοςΑν
ΤέλοςΕπανάληψης
Αν $x \bmod 2 = 1$ τότε
Γράψε '&'
ΤέλοςΑν
Γράψε '\$'
ΤέλοςΕπανάληψης
Γράψε '%'
για z από 1 μέχρι x+y
Γράψε '*'
ΤέλοςΕπανάληψης
για w από y μέχρι x
Γράψε '~'
ΤέλοςΕπανάληψης

Πόσες φορές θα εμφανισθούν οι χαρακτήρες: '!', '@', '#', '\$', '%', '^', '&', '*', '~';

```
Αλγόριθμος Άσκηση
α ← 1
Για i από 52 μέχρι 31
  με_βήμα-10
  β ← i - 11
  γ ← 2 * β
  Αν α > 15 τότε
    γ ← γ + α * β
  Αλλιώς
    γ ← γ div 3
  Τέλος_αν
  α ← β - γ
Τέλοςεπανάληψης
Εκτύπωσε α, β, γ
Τέλος Άσκηση
```

Θα εμφανισθούν οι τιμές: 7, 21, 14

```
Αλγόριθμος Άσκηση
β ← 10
Όσο β >= 0 επανάλαβε
  α ← β + 3
  Αν α < 8 τότε
    γ ← α - β
  Αλλιώς
    γ ← α + β
  Τέλος_αν
  Εκτύπωσε γ
  β ← β - 4
Τέλοςεπανάληψης
Τέλος Άσκηση
```

Θα εμφανισθούν οι τιμές: 23, 15, 3

*! Τιμές εισόδου:
! 5, 12, 9, 8, 21, 19, 0*


```
S1 ← 0
S2 ← 0
ΑρχήΕπανάληψης
  Διάβασε x
  Αν  $x \bmod 3 = 0$  τότε
    S1 ← S1 + x
  Αλλιώς
    S2 ← S2 + x
ΤέλοςΑν
ΜέχριςΌτου x = 0
Γράψε S1, S2
```

Θα εμφανισθούν οι τιμές: 42, 32

M10. Εμφωλευμένες επαναλήψεις, για-για

Πρόγραμμα που για καθένα από 25 αυτοκίνητα, διαβάζει: μοντέλο και 10 ποσοτικούς δείκτες. Εμφανίζει το μοντέλο με το μεγαλύτερο σύνολο δεικτών (χωρίς ισοτιμία)

```
max ← 0
για i από 1 μέχρι 25
  Διάβασε μ
  Σ ← 0
  για j από 1 μέχρι 10
    Διάβασε β
    Σ ← Σ + β
  ΤέλοςΕπανάληψης
  Αν (Σ > max) τότε
    max ← Σ
  maxM ← μ
  ΤέλοςΑν
ΤέλοςΕπανάληψης
Γράψε maxM
```



M10. Εμφωλευμένες επαναλήψεις, για-όσο/μέχρι

Στους προκριματικούς αγώνες ακοντισμού συμμετέχουν 25 αθλητές. Πρόγραμμα που για κάθε αθλητή, διαβάζει το όνομά του και τις ρίψεις του, μέχρι να γίνουν 5 ή μέχρι μία ρίψη να ξεπεράσει το όριο πρόκρισης (75m). Εμφανίζει ποιοί και πόσοι προκρίνονται και ποιά η μέση επίδοση αυτών που προκρίνονται.

```
Σ ← 0 Π ← 0
για i από 1 μέχρι 25
  Διάβασε ον
  ρίψεις ← 0
  ΑρχήΕπανάληψης
  Διάβασε ρ
  ρίψεις ← ρίψεις + 1
  ΜέχριςΌτου (ρ > 75 Η ρίψεις = 5)
  Αν (ρ > 75) τότε
    Γράψε "Προκρίνεται: ", ον
    Σ ← Σ + ρ
    Π ← Π + 1
  ΤέλοςΑν
ΤέλοςΕπανάληψης
Αν (Π <> 0) τότε
  Γράψε "Προκρίνονται: ", Π, " αθλητές"
  Γράψε "Μέση επίδοση: ", Σ/Π, " m"
Αλλιώς
  Γράψε "Κανείς δεν προκρίθηκε"
ΤέλοςΑν
```

M10. Εμφωλευμένες επαναλήψεις, όσο/μέχρι - για

Πρόγραμμα που διαβάζει επαναληπτικά για κάθε μαθητή: όνομα και βαθμούς σε 10 μαθήματα, μέχρι να δοθεί για όνομα το κενό (""). Εμφανίζει για κάθε μαθητή το μέσο όρο του και τελικά το όνομα του μαθητή με τον μεγαλύτερο μέσο όρο (χωρίς ισοτιμία).

$max \leftarrow 0$

Διάβασε ον

Όσο (ον <> "") επανάλαβε

$\Sigma \leftarrow 0$

για i από 1 μέχρι 10

Διάβασε β

$\Sigma \leftarrow \Sigma + \beta$

ΤέλοςΕπανάληψης

$\mu o \leftarrow \Sigma / 10$

Γράψε μο

Αν ($\mu o > max$) τότε

$max \leftarrow \mu o$

$maxO n \leftarrow \text{ον}$

ΤέλοςΑν

Διάβασε ον

ΤέλοςΕπανάληψης


Αν ($max <> 0$) τότε

Γράψε $maxO n$

Αλλιώς

Γράψε "Δεν δόθηκαν στοιχεία"

ΤέλοςΑν



M10. Εμφωλευμένες επαναλήψεις, όσο/μέχρι - όσο/μέχρι

Πρόγραμμα που διαβάζει επαναληπτικά για κάθε πωλητή: όνομα (μέχρι το κενό "") και πωλήσεις (μέχρι να δοθεί για πώληση το -1). Εμφανίζει για κάθε πωλητή το σύνολο των πωλήσεών του.

Διάβασε ον

Όσο (ον <> "") επανάλαβε

$\Sigma \leftarrow 0$

Διάβασε π

Όσο ($\pi <> -1$) επανάλαβε

$\Sigma \leftarrow \Sigma + \pi$

Διάβασε π

ΤέλοςΕπανάληψης

Γράψε Σ

Διάβασε ον

ΤέλοςΕπανάληψης

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M11. Έλεγχος συνόλου **γνωστού** πλήθους στοιχείων εάν όλα τα στοιχεία του πληρούν μία ιδιότητα

π.χ. Πρόγραμμα που διαβάζει 100 θετικούς ακέραιους και ελέγχει εάν και οι 100 ήταν άρτιοι.

για i από 1 μέχρι 100
Διάβασε a
Αν $(a \bmod 2 = 0)$ τότε
 Γράψε 'ΝΑΙ'
Αλλιώς
 Γράψε 'ΟΧΙ'
ΤέλοςΑν
ΤέλοςΕπανάληψης
! Είναι σωστό;

για i από 1 μέχρι 100
Διάβασε a
Αν $(a \bmod 2 = 0)$ τότε
 άρτιοι \leftarrow Αληθής
Αλλιώς
 άρτιοι \leftarrow Ψευδής
ΤέλοςΑν
ΤέλοςΕπανάληψης
Γράψε άρτιοι
! Είναι σωστό;

1ος τρόπος

```
Π ← 0
για i από 1 μέχρι 100
  Διάβασε a
  Αν (a mod 2 = 0) τότε
    Π ← Π + 1
  ΤέλοςΑν
ΤέλοςΕπανάληψης
Αν (Π = 100) τότε
  Γράψε 'ΝΑΙ'
Αλλιώς
  Γράψε 'ΟΧΙ'
ΤέλοςΑν
```

2ος τρόπος

```
άρτιοι ← Αληθής
για i από 1 μέχρι 100
  Διάβασε a
  Αν (a mod 2 <> 0) τότε
    άρτιοι ← Ψευδής
  ΤέλοςΑν
ΤέλοςΕπανάληψης
Αν (άρτιοι = Αληθής) τότε
  Γράψε 'ΝΑΙ'
Αλλιώς
  Γράψε 'ΟΧΙ'
ΤέλοςΑν
```

3ος τρόπος (πιο αποδοτικός)

```
i ← 1
άρτιοι ← Αληθής
Όσο (i <= 100 ΚΑΙ άρτιοι = Αληθής) επανάλαβε
  Διάβασε a
  Αν (a mod 2 <> 0) τότε
    άρτιοι ← Ψευδής
  ΤέλοςΑν
  i ← i + 1
ΤέλοςΕπανάληψης
Αν (άρτιοι = Αληθής) τότε
  Γράψε 'ΝΑΙ'
Αλλιώς
  Γράψε 'ΟΧΙ'
ΤέλοςΑν
```

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M11. Έλεγχος συνόλου αγνώστου πλήθους στοιχείων εάν όλα τα στοιχεία του πληρούν μία ιδιότητα

Εισαγωγή θετικών ακέραιων μέχρι να δοθεί το 0. Ήταν όλοι άρτιοι; (θα δοθεί τουλάχιστον ένας θετικός)

Εισαγωγή των τιμών "κλεισίματος" μίας μετοχής (θετικές) μέχρι να δοθεί το 0. Ήταν συνεχώς ανοδική η τιμή της;

$\Pi \leftarrow 0$ $\Pi_A \leftarrow 0$

Διάβασε x

Όσο $x \neq 0$ επανάλαβε

$\Pi \leftarrow \Pi + 1$

Αν $x \bmod 2 = 0$ τότε

$\Pi_A \leftarrow \Pi_A + 1$

ΤέλοςΑν

Διάβασε x

ΤέλοςΕπανάληψης

Αν $\Pi = \Pi_A$ τότε

Γράψε 'ναι'


Αλλιώς

Γράψε 'όχι'

ΤέλοςΑν

! ή υπολογισμός του πλήθους των περιπτώσεων (ΠΕΡ)

! Αν (ΠΕΡ = 0) τότε Γράψε 'ΝΑΙ'



$\Pi \leftarrow 0$

$\Pi_A \leftarrow 0$

$\text{pr} \leftarrow 0$! προηγούμενη τιμή

Διάβασε x

Όσο $x \neq 0$ επανάλαβε

$\Pi \leftarrow \Pi + 1$

Αν $x > \text{pr}$ τότε

$\Pi_A \leftarrow \Pi_A + 1$

ΤέλοςΑν

$\text{pr} \leftarrow x$

Διάβασε x

ΤέλοςΕπανάληψης

Αν $\Pi_A = \Pi$ τότε

Γράψε 'ναι'

Αλλιώς

Γράψε 'όχι'

ΤέλοςΑν

M12. Εξάντληση αποθέματος

Ένας μανάβης προμηθεύεται 300 κιλά μήλα για να τα πουλήσει με 2€ το κιλό. Να γραφεί πρόγραμμα το οποίο για κάθε παραγγελία διαβάσει την ποσότητα σε κιλά και εφόσον το απόθεμα επαρκεί για την κάλυψη της παραγγελίας να την εκτελεί. Διαφορετικά να διαθέτει το διαθέσιμο απόθεμα. Η εισαγωγή τερματίζεται όταν εξαντληθεί το απόθεμα. Για κάθε παραγγελία να εμφανίζει το κόστος της.

```
ΑΠ ← 300
Όσο (ΑΠ <> 0) επανάλαβε
  Διάβασε ποσ
  Αν (ποσ > ΑΠ) τότε
    ποσ ← ΑΠ
  ΤέλοςΑν
  ΑΠ ← ΑΠ - ποσ
  Γράψε ποσ*2, " €"
ΤέλοςΕπανάληψης
```

M13. Αποφυγή παραβίασης άνω / κάτω φράγματος

π.χ. Ένα ασανσέρ έχει μέγιστο όριο φορτίου τα 350 κιλά. Πρόγραμμα που δέχεται επαναληπτικά το βάρος του κάθε ατόμου που εισέρχεται σε αυτό μέχρι να παραβιασθεί αυτό το όριο. Εμφανίζει τελικά το πλήθος των ατόμων που επιτρέπεται να εισέλθουν καθώς και το συνολικό τους βάρος.

1ος τρόπος

```
Π ← 0
SB ← 0
Διάβασε β
Όσο (SB + β <= 350) επανάλαβε
  SB ← SB + β
  Π ← Π + 1
Διάβασε β
ΤέλοςΕπανάληψης
Γράψε Π, SB
```

2ος τρόπος

```
Π ← 0
Υπ ← 350 ! υπόλοιπο
Διάβασε β
Όσο (Υπ >= β) επανάλαβε
  Υπ ← Υπ - β
  Π ← Π + 1
Διάβασε β
ΤέλοςΕπανάληψης
SB ← 350 - Υπ
Γράψε Π, SB
```

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

1567 4 digits	256 3 digits	58964 5 digits
------------------	-----------------	-------------------

M14. Ανάλυση ακέραιου στα ψηφία του

π.χ. Εισαγωγή θετικού ακέραιου και εμφάνιση του μέσου όρου των ψηφίων του. Π.χ. με είσοδο την τιμή 1234 να βγάξει έξοδο την τιμή 2.5

ΠΡΟΓΡΑΜΜΑ mesos_oros

ΜΕΤΑΒΛΗΤΕΣ

ΑΚΕΡΑΙΕΣ: x, S, p

ΑΡΧΗ

ΓΡΑΨΕ 'Δώσε θετικό ακέραιο'

ΔΙΑΒΑΣΕ x

S ← 0

p ← 0

ΟΣΟ x <> 0 **ΕΠΑΝΑΛΑΒΕ**

p ← p + 1

S ← S + x mod 10

x ← x div 10

ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

ΓΡΑΨΕ S / p

ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ

! εμφάνιση των ψηφίων με τη σειρά

ΔΙΑΒΑΣΕ x

p ← 1

ΟΣΟ x div 10^p <> 0 ΕΠΑΝΑΛΑΒΕ

p ← p + 1

ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

ΓΙΑ i ΑΠΟ p-1 ΜΕΧΡΙ 0 ΜΕ ΒΗΜΑ -1

ΓΡΑΨΕ x div 10ⁱ mod 10

ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

123

1

2

3

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M15. Γνωστό μέγιστο πλήθος επαναλήψεων (Όσο/Μέχρι)

π.χ. Σε ένα τηλεπαιχνίδι ο παίκτης ξεκινάει με 300000€ και καλείται να απαντήσει σε 7 ερωτήσεις πολλαπλής επιλογής. Σε κάθε ερώτηση ο παίκτης μπορεί να ποντάρει τα χρήματα που διαθέτει στις προτεινόμενες απαντήσεις. Προχωράει στην επόμενη ερώτηση με τα χρήματα που πόνταρε στη σωστή απάντηση. Σταματάει το παιχνίδι εάν μείνει χωρίς χρήματα ή εφόσον τελειώσει και την έβδομη ερώτηση. Πρόγραμμα που προσομοιώνει το παιχνίδι ως εξής: σε κάθε ερώτηση και εφόσον ο παίκτης έχει ακόμη χρήματα, να διαβάσει το ποσό που πόνταρε στη σωστή απάντηση με έλεγχο εγκυρότητας ώστε να είναι το πολύ όσο το και το ποσό που του έχει μείνει. Τελικά να εμφανίζει τα χρήματα που κέρδισε ή την ερώτηση (1-7) στην οποία τα έχασε όλα.

$x \leftarrow 300000$ $ερ \leftarrow 0$

Όσο ($ερ < 7$ ΚΑΙ $x <> 0$) επανάλαβε

$ερ \leftarrow ερ + 1$

ΑρχήΕπανάληψης

Διάβασε ποσό

ΜέχριςΌτου ($ποσό \geq 0$ ΚΑΙ $ποσό \leq x$)

$x \leftarrow ποσό$

ΤέλοςΕπανάληψης

Αν ($x <> 0$) τότε

Γράψε 'Κέρδισε ', x , '€'

Αλλιώς


Γράψε 'Έχασε στην ερώτηση: ', $ερ$

ΤέλοςΑν



(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M16. Καταγραφή των στοιχείων ενός συνόλου που πληρούν μία ιδιότητα



όλα τα στοιχεία $\Leftrightarrow \Pi = N$

κάποια από τα στοιχεία $\Leftrightarrow \Pi <> 0$

κανένα από τα στοιχεία $\Leftrightarrow \Pi = 0$

% ποσοστό στοιχείων $\Leftrightarrow \Pi/N*100$

μέγιστο σερί στοιχείων

π.χ. πρόγραμμα που διαβάζει 100 θετικές ακέραιες τιμές και εμφανίζει το μέγιστο **σερί των άρτιων**

```

maxΣερί <-- 0
σερί <-- 0
για i από 1 μέχρι 100
  Διάβασε x
  Αν x mod 2 = 0 τότε
    σερί <-- σερί + 1
  Αλλιώς
    σερί <-- 0
  ΤέλοςΑν
  Αν (σερί > maxΣερί) τότε
    maxΣερί <-- σερί
  ΤέλοςΑν
ΤέλοςΕπανάληψης
Γράψε maxΣερί

```

μέγιστο σερί στοιχείων

π.χ. πρόγραμμα που διαβάζει 100 ακέραιες τιμές και εμφανίζει το μέγιστο **σερί των ίδιων τιμών**

```

maxΣερί <-- 1
σερί <-- 1
Διάβασε πρ
για i από 2 μέχρι 100
  Διάβασε x
  Αν x = πρ τότε
    σερί <-- σερί + 1
  Αλλιώς
    σερί <-- 0
  ΤέλοςΑν
  Αν (σερί > maxΣερί) τότε
    maxΣερί <-- σερί
  ΤέλοςΑν
  πρ <-- x
ΤέλοςΕπανάληψης
Γράψε maxΣερί

```

M17. Παιχνίδι με 2 παίκτες

(π.χ. σε κάθε γύρο ρίχνουν 1 ζάρι και κερδίζει αυτός που έφερε το μεγαλύτερο)



α) τελειώνει στους N γύρους

$$\text{νικές}_1 \leftarrow 0 \quad \text{νικές}_2 \leftarrow 0$$

για γύρο από 1 μέχρι N

! έστω ζ_1, ζ_2 οι ζαριές

Αν $(\zeta_1 > \zeta_2)$ τότε

$$\text{νικές}_1 \leftarrow \text{νικές}_1 + 1$$

Αλλιώς Αν $(\zeta_2 > \zeta_1)$ τότε

$$\text{νικές}_2 \leftarrow \text{νικές}_2 + 1$$

Τέλος Αν

Τέλος Επανάληψης

- $\text{νικές}_1 > \text{νικές}_2 \Rightarrow$ τελικός νικητής ο 1^{ος}
- $\text{νικές}_2 > \text{νικές}_1 \Rightarrow$ τελικός νικητής ο 2^{ος}
- $\text{νικές}_1 = \text{νικές}_2 \Rightarrow$ Ισοπαλία

γ) τελειώνει στις N σερί νίκες

$$\text{σερί}_1 \leftarrow 0 \quad \text{σερί}_2 \leftarrow 0$$

Όσο $(\text{σερί}_1 < N \text{ ΚΑΙ } \text{σερί}_2 < N)$ επανάλαβε

! ή ... Μέχρις Ότου $(\text{σερί}_1 = N \text{ Η } \text{σερί}_2 = N)$

! έστω ζ_1, ζ_2 οι ζαριές

Αν $(\zeta_1 > \zeta_2)$ τότε

$$\text{σερί}_1 \leftarrow \text{σερί}_1 + 1$$

$$\text{σερί}_2 \leftarrow 0$$

Αλλιώς Αν $(\zeta_2 > \zeta_1)$ τότε

$$\text{σερί}_2 \leftarrow \text{σερί}_2 + 1$$

$$\text{σερί}_1 \leftarrow 0$$

Τέλος Αν

Τέλος Επανάληψης

β) τελειώνει στις N νίκες

$$\text{νικές}_1 \leftarrow 0$$

$$\text{νικές}_2 \leftarrow 0$$

Όσο $(\text{νικές}_1 < N \text{ ΚΑΙ } \text{νικές}_2 < N)$ επανάλαβε

! ή ... Μέχρις Ότου $(\text{νικές}_1 = N \text{ Η } \text{νικές}_2 = N)$

! έστω ζ_1, ζ_2 οι ζαριές

Αν $(\zeta_1 > \zeta_2)$ τότε

$$\text{νικές}_1 \leftarrow \text{νικές}_1 + 1$$

Αλλιώς Αν $(\zeta_2 > \zeta_1)$ τότε

$$\text{νικές}_2 \leftarrow \text{νικές}_2 + 1$$

Τέλος Αν

Τέλος Επανάληψης

- $\text{νικές}_1 = N \Rightarrow$ τελικός νικητής ο 1^{ος}
- $\text{νικές}_2 = N \Rightarrow$ τελικός νικητής ο 2^{ος}

$$\text{σερί}_1 = N \Rightarrow \text{τελικός νικητής ο 1ος}$$

$$\text{σερί}_2 = N \Rightarrow \text{τελικός νικητής ο 2ος}$$

M18. Πίνακας τιμών με γνωστή έξοδο και άγνωστο την είσοδο

Ποια τιμή πρέπει να εισαχθεί στο A ώστε να εμφανισθεί η τιμή -19;

ΠΡΟΓΡΑΜΜΑ M18

ΜΕΤΑΒΛΗΤΕΣ

ΑΚΕΡΑΙΕΣ: A, x, S

ΑΡΧΗ

ΔΙΑΒΑΣΕ A

$S \leftarrow 0$

ΓΙΑ x ΑΠΟ 1 ΜΕΧΡΙ A

ΑΝ $x \bmod 3 = 0$ ΤΟΤΕ

$S \leftarrow S + x$

ΑΛΛΙΩΣ

$S \leftarrow S - x$

ΤΕΛΟΣ_ΑΝ

ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

ΓΡΑΨΕ S

ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ

x	S
	0
1	-1
2	-3
3	0
4	-4
5	-9
6	-3
7	-10
8	-18
9	-9
10	-19

Απάντηση: A = 10

M19. "Μοτίβα" ακολουθίας αριθμών

α. Εμφάνιση των τιμών:

1,3,5,7,9,8,10,12,14,16,15,17,19,21,23,22,24,26,28,30

ΓΙΑ x ΑΠΟ 1 ΜΕΧΡΙ 22 ΜΕ ΒΗΜΑ 7

ΓΙΑ y ΑΠΟ x ΜΕΧΡΙ x+8 ΜΕ ΒΗΜΑ 2

ΓΡΑΨΕ y

ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

1,3,5,7,9,8,10,12,14,16,15,17,19,21,23,22,24,26,28,30

β. Συμπληρώστε τα κενά 1-7 ώστε να εμφανισθούν οι τιμές:

5,7,9,11,10,12,14,16,15,17,19,21

ΓΙΑ x ΑΠΟ ..(1).. ΜΕΧΡΙ ..(2).. ΜΕ ΒΗΜΑ ..(3)..

ΓΙΑ y ΑΠΟ ..(4).. ΜΕΧΡΙ ..(5).. ΜΕ ΒΗΜΑ ..(6)..

ΓΡΑΨΕ ..(7)..

ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

5,7,9,11, 10,12,14,16, 15,17,19,21

(1) 5, (2) 15, (3) 5, (4) x, (5) x+6, (6) 2, (7) y

ή

(1) 1, (2) 3, (3) 1, (4) 5*x, (5) 5*x+6, (6) 2, (7) y

ή

(1) 5, (2) 15, (3) 0, (4) 6, (5) 2, (6) 2, (7) x + y

ή

(1) 5, (2) 15, (3) 5, (4) 1, (5) 4, (6) 1, (7) x+(y-1)*2

M20. Πρώτοι αριθμοί (διαιρούνται μόνο με το 1 και τον εαυτό τους π.χ. το 7) και τέλειοι αριθμοί (έχουν άθροισμα διαιρετών ίσο με το 2πλάσιο του εαυτού τους. π.χ. το 6)

Διάβασε x ! *έλεγχος για πρώτος*

$pd \leftarrow 0$

για y από 1 μέχρι x

Αν $(x \bmod y = 0)$ τότε

$pd \leftarrow pd + 1$

ΤέλοςΑν

ΤέλοςΕπανάληψης

Γράψε ($pd = 2$)

Διάβασε x ! *έλεγχος για τέλειος*

$sd \leftarrow 0$

για y από 1 μέχρι x

Αν $(x \bmod y = 0)$ τότε

$sd \leftarrow sd + y$

ΤέλοςΑν

ΤέλοςΕπανάληψης


Γράψε ($sd = 2*x$)

Ανασκόπηση

Δομή Επανάληψης	Επαναληπτικά προβλήματα	
	Γνωστού πλήθους επαναλήψεων	Αγνώστου πλήθους επαναλήψεων
Όσο	✓	✓
Μέχρι	✓	✓
Για	✓ (προτιμότερη)	✗

(§ 8.2.3) Κανόνες χρήσης των εμφωλευμένων βρόχων

Βρόχος: τμήμα εντολών που εσωκλείεται σε μία δομή επανάληψης



Κανόνες χρήσης των εμφωλευμένων βρόχων:

1. Ο εσωτερικός βρόχος πρέπει να βρίσκεται ολόκληρος μέσα στον εξωτερικό. Ο βρόχος που ξεκινάει τελευταίος, πρέπει να ολοκληρώνεται πρώτος.
2. Η είσοδος σε κάθε βρόχο υποχρεωτικά γίνεται από την αρχή του.
3. Δεν μπορεί να χρησιμοποιηθεί η ίδια μεταβλητή ως μετρητής δύο ή περισσότερων βρόχων που ο ένας βρίσκεται στο εσωτερικό του άλλου

M21. Ερμηνεία κώδικα

Δίνεται το παρακάτω πρόγραμμα. Τι εμφανίζει στα σημεία (1) - (7);

```
S <-- 0 Π <-- 0
SA <-- 0 ΠΑ <-- 0 maxA <-- 0
ΣΓ <-- 0 ΠΓ <-- 0 ΠΓΚ <-- 0
max <-- 0 Πmax <-- 0 f700 <-- Ψευδής
Διάβασε on
'Όσο on <> " επανάλαβε
  ΑρχήΕπανάληψης
  Διάβασε φ
  Μέχρις'Ότου φ = 'Α' Η φ = 'Γ'
  Διάβασε μισθός
  S <-- S + μισθός
  Π <-- Π + 1
  Αν φ = 'Α' τότε
    SA <-- SA + μισθός
    ΠΑ <-- ΠΑ + 1
    Αν μισθός > maxA τότε
      maxA <-- μισθός
    ΤέλοςΑν
  Αλλιώς
    ΣΓ <-- ΣΓ + μισθός
    ΠΓ <-- ΠΓ + 1
    Αν μισθός < 1000 τότε
      ΠΓΚ <-- ΠΓΚ + 1
    ΤέλοςΑν
  ΤέλοςΑν
```

```
Αν μισθός > max τότε
  max <-- μισθός
  Πmax <-- 1
ΑλλιώςΑν μισθός = max τότε
  Πmax <-- Πmax + 1
ΤέλοςΑν
Αν μισθός < 700 τότε
  f700 <-- Αληθής
ΤέλοςΑν
Διάβασε on
ΤέλοςΕπανάληψης
Αν Π <> 0 τότε
  Γράψε S / Π ! (1)
Αν ΠΑ <> 0 τότε
  Γράψε SA / ΠΑ ! (2)
  Γράψε maxA ! (4)
ΤέλοςΑν
Αν ΠΓ <> 0 τότε
  Γράψε ΣΓ / ΠΓ ! (3)
  Γράψε ΠΓΚ / ΠΓ * 100, '%' ! (5)
ΤέλοςΑν
Γράψε max, Πmax ! (6)
Αν f700 τότε ! (7)
  Γράψε 'ναι'
Αλλιώς
  Γράψε 'όχι'
ΤέλοςΑν
ΤέλοςΑν
```

Απάντηση:

- (1) τον μέσο όρο των μισθών
- (2) τον μέσο όρο των μισθών των αντρών
- (3) τον μέσο όρο των μισθών των γυναικών
- (4) τον μέγιστο μισθό των αντρών
- (5) το % ποσοστό των γυναικών με μισθό κάτω των 1000€ (στο σύνολο των γυναικών)
- (6) τον μέγιστο μισθό και το πλήθος των υπαλλήλων που τον λαμβάνουν
- (7) κατάλληλο μήνυμα εάν υπάρχει μισθός κάτω των 700€

M22. Πρώτο/Τελευταίο εισαγόμενο στοιχείο με μία ιδιότητα

α. γνωστού πλήθους εισαγωγές
π.χ. Εισαγωγή 100 ακέραιων τιμών. Εμφάνιση του πρώτου και του τελευταίου άρτιου.

```
ρΑ <-- 0
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 100
  ΔΙΑΒΑΣΕ x
  ΑΝ x mod 2 = 0 ΤΟΤΕ
    ρΑ <-- ρΑ + 1
    ΑΝ ρΑ = 1 ΤΟΤΕ
      first <-- x
    ΤΕΛΟΣ_ΑΝ
  last <-- x
ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΑΝ ρΑ <> 0 ΤΟΤΕ
  ΓΡΑΨΕ 'first=',first,'last=',last
ΑΛΛΙΩΣ
  ΓΡΑΨΕ 'kanenas'
ΤΕΛΟΣ_ΑΝ
```

β. αγνώστου πλήθους εισαγωγές
π.χ. Εισαγωγή ακέραιων τιμών μέχρι να δοθεί το -1.
Εμφάνιση του πρώτου και του τελευταίου άρτιου.

```
ρΑ <-- 0
ΔΙΑΒΑΣΕ x
ΟΣΟ x <> -1 ΕΠΑΝΑΛΑΒΕ
  ΑΝ x mod 2 = 0 ΤΟΤΕ
    ρΑ <-- ρΑ + 1
    ΑΝ ρΑ = 1 ΤΟΤΕ
      first <-- x
    ΤΕΛΟΣ_ΑΝ
  last <-- x
ΤΕΛΟΣ_ΑΝ
ΔΙΑΒΑΣΕ x
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΑΝ ρΑ <> 0 ΤΟΤΕ
  ΓΡΑΨΕ 'first=',first,'last=',last
ΑΛΛΙΩΣ
  ΓΡΑΨΕ 'kanenas'
ΤΕΛΟΣ_ΑΝ
```

M23. Ισοδυναμία αλγορίθμων

Συμπληρώσετε τα κενά ώστε τα ακόλουθα τμήματα αλγορίθμων να έχουν την ίδια έξοδο:

```
p <-- 0
για i από 1 μέχρι 3
  Διάβασε x
  Αν  $x \bmod 2 = 1$  τότε
    p <-- p + 1
  ΤέλοςΑν
ΤέλοςΕπανάληψης
Γράψε p
```

```
Διάβασε x1, x2, x3
p <-- .....(1).....
Γράψε p
```

```
p <-- 0
για i από 1 μέχρι 3
  Διάβασε x
  p <-- .....(2).....
ΤέλοςΕπανάληψης
Γράψε p
```

```
(1)  $x1 \bmod 2 + x2 \bmod 2 + x3 \bmod 2$   
(2)  $p + x \bmod 2$ 
```

M24. Διερεύνηση εξίσωσης - συνδυασμοί τιμών



Πρόγραμμα που βρίσκει ποιές και πόσες είναι οι ακέραιες ρίζες της εξίσωσης: $3x - 7y + 5z = 11$ για κάθε ακέραια τιμή των x, y, z στο διάστημα $[-100, 100]$

```
p <-- 0
για x από -100 μέχρι 100
  για y από -100 μέχρι 100
    για z από -100 μέχρι 100
      Αν  $3x - 7y + 5z = 11$  τότε
        Γράψε x, y, z
      p <-- p + 1
    ΤέλοςΑν
  ΤέλοςΕπανάληψης
ΤέλοςΕπανάληψης
ΤέλοςΕπανάληψης
Γράψε p
```

(§ 2.4.5) Πολλαπλασιασμός αλά ρωσικά

$\gamma \leftarrow 0$
Διάβασε $a, \beta \neq 0$
Όσο ($\beta > 0$) επανάλαβε
 Αν ($\beta \bmod 2 \neq 0$) τότε
 $\gamma \leftarrow \gamma + a$
 ΤέλοςΑν
 $a \leftarrow a * 2$
 $\beta \leftarrow \beta \text{ div } 2$
ΤέλοςΕπανάληψης
Γράψε γ

45	19	45
90	9	90
180	4	
360	2	
720	1	720
Άθροισμα = 855		



Ολίσθηση (shift)

Στα κυκλώματα του υπολογιστή τα δεδομένα αποθηκεύονται με δυαδική μορφή, δηλαδή 0 και 1, ανεξάρτητα από το πως τα ορίζει ο προγραμματιστής, όπως ακέραιους ή πραγματικούς σε δεκαδικό σύστημα, ή ακόμη χαρακτήρες κ.λπ. Έτσι ο αριθμός 17 του δεκαδικού συστήματος ισοδυναμεί με τον αριθμό 00010001 του δυαδικού συστήματος, ο οποίος μπορεί να αποθηκευθεί σε ένα byte. Αν μετακινήσουμε τα ψηφία αυτά κατά μία θέση προς τα αριστερά, δηλαδή αν προσθέσουμε ένα 0 στο τέλος του αριθμού και αγνοήσουμε το αρχικό 0, τότε προκύπτει ο αριθμός 00100010 του δυαδικού συστήματος, που ισοδυναμεί με το αριθμό 34 του δεκαδικού συστήματος. Επίσης, με παρόμοιο τρόπο, αν μετακινήσουμε τα ψηφία κατά μία θέση δεξιά, δηλαδή αποκόψουμε το τελευταίο ψηφίο 1 και θεωρήσουμε ένα ακόμη αρχικό 0, τότε προκύπτει ο αριθμός 00001000 του δυαδικού συστήματος, που ισοδυναμεί με τον αριθμό 8 του δεκαδικού συστήματος. Άρα η ολίσθηση προς τα αριστερά ισοδυναμεί με πολλαπλασιασμό επί δύο, ενώ η ολίσθηση προς τα δεξιά ισοδυναμεί με την ακέραια διαίρεση διά δύο.